Decomposition of totally ordered hoops

Sebastián Buss

April 18, 2024

Abstract

We formalize a well known result in theory of hoops: every totally ordered hoop can be written as an ordinal sum of irreducible (equivalently Wajsberg) hoops. This formalization is based on the proof for BL-chains (i.e., bounded totally ordered hoops) by Busaniche [5].

Contents

1 Some order tools: posets with explicit universe

theory *Posets*
imports *Main HOL−Library.LaTeXsugar*

begin

locale *poset-on* =
  fixes P :: $'b$ *set*
  fixes *P-lesseq* :: $'b \Rightarrow 'b \Rightarrow$ *bool* (**infix** $\leq^P$ *60*)
  fixes *P-less* :: $'b \Rightarrow 'b \Rightarrow$ *bool* (**infix** $<^P$ *60*)
  assumes *not-empty* [*simp*]: $P \neq \emptyset$
  and *reflex*: *reflp-on* P $(\leq^P)$
  and *antisymm*: *antisymp-on* P $(\leq^P)$
  and *trans*: *transp-on* P $(\leq^P)$
  and *strict-iff-order*: $x \in P \Longrightarrow y \in P \Longrightarrow x <^P y = (x \leq^P y \wedge x \neq y)$
begin

lemma *strict-trans*:
  assumes $a \in P$ $b \in P$ $c \in P$ $a <^P b$ $b <^P c$
  shows $a <^P c$
  using *antisymm antisymp-onD assms trans strict-iff-order transp-onD*
  by (*smt* (*verit*, *ccfv-SIG*))

end

locale *bot-poset-on* = *poset-on* +
  fixes *bot* :: $'b$ (0^P)
  assumes *bot-closed*: $0^P \in P$
  and *bot-first*: $x \in P \Longrightarrow 0^P \leq^P x$

locale *top-poset-on* = *poset-on* +
  fixes *top* :: $'b$ (1^P)
  assumes *top-closed*: $1^P \in P$
  and *top-last*: $x \in P \Longrightarrow x \leq^P 1^P$

locale *bounded-poset-on* = *bot-poset-on* + *top-poset-on*

locale *total-poset-on* = *poset-on* +
  assumes *total*: *totalp-on* P $(\leq^P)$
begin

lemma *trichotomy*:
  assumes $a \in P$ $b \in P$
  shows $(a <^P b \wedge \neg(a = b \vee b <^P a)) \vee$
        $(a = b \wedge \neg(a <^P b \vee b <^P a)) \vee$
        $(b <^P a \wedge \neg(a = b \vee a <^P b))$
  using *antisymm antisymp-onD assms strict-iff-order total totalp-onD* **by** *metis*

lemma *strict-order-equiv-not-converse*:
  assumes $a \in P$ $b \in P$
  shows $a <^P b \longleftrightarrow \neg(b \leq^P a)$
  using *assms strict-iff-order reflex reflp-onD strict-trans trichotomy* **by** *metis*

end

end

2 Hoops

A *hoop* is a naturally ordered *pocrim* (i.e., a partially ordered commutative residuated integral monoid). This structures have been introduced by Büchi and Owens in [4] and constitute the algebraic counterpart of fragments without negation and falsum of some nonclassical logics.

theory *Hoops*
  imports *Posets*
begin

2.1 Definitions

locale *hoop* =
  fixes *universe* :: $'a$ *set* (A)
  and *multiplication* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infix** $*^A$ *60*)
  and *implication* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infix** $\rightarrow^A$ *60*)
  and *one* :: $'a$ (1^A)
  assumes *mult-closed*: $x \in A \Longrightarrow y \in A \Longrightarrow x *^A y \in A$
  and *imp-closed*: $x \in A \Longrightarrow y \in A \Longrightarrow x \rightarrow^A y \in A$
  and *one-closed* [*simp*]: $1^A \in A$
  and *mult-comm*: $x \in A \Longrightarrow y \in A \Longrightarrow x *^A y = y *^A x$
  and *mult-assoc*: $x \in A \Longrightarrow y \in A \Longrightarrow z \in A \Longrightarrow x *^A (y *^A z) = (x *^A y) *^A z$
  and *mult-neutr* [*simp*]: $x \in A \Longrightarrow x *^A 1^A = x$
  and *imp-reflex* [*simp*]: $x \in A \Longrightarrow x \rightarrow^A x = 1^A$
  and *divisibility*: $x \in A \Longrightarrow y \in A \Longrightarrow x *^A (x \rightarrow^A y) = y *^A (y \rightarrow^A x)$
  and *residuation*: $x \in A \Longrightarrow y \in A \Longrightarrow z \in A \Longrightarrow$
        $x \rightarrow^A (y \rightarrow^A z) = (x *^A y) \rightarrow^A z$
begin

definition *hoop-order* :: $'a \Rightarrow 'a \Rightarrow bool$ (**infix** $\leq^A$ *60*)
  where $x \leq^A y \equiv (x \rightarrow^A y = 1^A)$

definition *hoop-order-strict* :: $'a \Rightarrow 'a \Rightarrow bool$ (**infix** $<^A$ *60*)
  where $x <^A y \equiv (x \leq^A y \wedge x \neq y)$

definition *hoop-inf* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infix** $\wedge^A$ *60*)
  where $x \wedge^A y = x *^A (x \rightarrow^A y)$

definition *hoop-pseudo-sup* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infix** $\vee^{*A}$ *60*)

where $x \vee^{*A} y = ((x \to^A y) \to^A y) \wedge^A ((y \to^A x) \to^A x)$

end

locale *wajsberg-hoop* = *hoop* +
assumes *T*: $x \in A \Longrightarrow y \in A \Longrightarrow (x \to^A y) \to^A y = (y \to^A x) \to^A x$
begin

definition *wajsberg-hoop-sup* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infix** $\vee^A$ *60*)
where $x \vee^A y = (x \to^A y) \to^A y$

end

2.2 Basic properties

context *hoop*
begin

lemma *mult-neutr-2* [*simp*]:
assumes $a \in A$
shows $1^A *^A a = a$
using *assms mult-comm* **by** *simp*

lemma *imp-one-A*:
assumes $a \in A$
shows $(1^A \to^A a) \to^A 1^A = 1^A$
proof −
have $(1^A \to^A a) \to^A 1^A = (1^A \to^A a) \to^A (1^A \to^A 1^A)$
using *assms* **by** *simp*
also
have $\ldots = ((1^A \to^A a) *^A 1^A) \to^A 1^A$
using *assms imp-closed residuation* **by** *simp*
also
have $\ldots = ((a \to^A 1^A) *^A a) \to^A 1^A$
using *assms divisibility imp-closed mult-comm* **by** *simp*
also
have $\ldots = (a \to^A 1^A) \to^A (a \to^A 1^A)$
using *assms imp-closed one-closed residuation* **by** *metis*
also
have $\ldots = 1^A$
using *assms imp-closed* **by** *simp*
finally
show *?thesis*
by *auto*
qed

lemma *imp-one-B*:
assumes $a \in A$
shows $(1^A \to^A a) \to^A a = 1^A$

proof −
  have $(1^A \rightarrow^A a) \rightarrow^A a = ((1^A \rightarrow^A a) *^A 1^A) \rightarrow^A a$
    using *assms imp-closed* **by** *simp*
  also
  have $\ldots = (1^A \rightarrow^A a) \rightarrow^A (1^A \rightarrow^A a)$
    using *assms imp-closed one-closed residuation* **by** *metis*
  also
  have $\ldots = 1^A$
    using *assms imp-closed* **by** *simp*
  finally
  show *?thesis*
    by *auto*
qed

lemma *imp-one-C*:
  assumes $a \in A$
  shows $1^A \rightarrow^A a = a$
proof −
  have $1^A \rightarrow^A a = (1^A \rightarrow^A a) *^A 1^A$
    using *assms imp-closed* **by** *simp*
  also
  have $\ldots = (1^A \rightarrow^A a) *^A ((1^A \rightarrow^A a) \rightarrow^A a)$
    using *assms imp-one-B* **by** *simp*
  also
  have $\ldots = a *^A (a \rightarrow^A (1^A \rightarrow^A a))$
    using *assms divisibility imp-closed* **by** *simp*
  also
  have $\ldots = a$
    using *assms residuation* **by** *simp*
  finally
  show *?thesis*
    by *auto*
qed

lemma *imp-one-top*:
  assumes $a \in A$
  shows $a \rightarrow^A 1^A = 1^A$
proof −
  have $a \rightarrow^A 1^A = (1^A \rightarrow^A a) \rightarrow^A 1^A$
    using *assms imp-one-C* **by** *auto*
  also
  have $\ldots = 1^A$
    using *assms imp-one-A* **by** *auto*
  finally
  show *?thesis*
    by *auto*
qed

The proofs of *imp-one-A*, *imp-one-B*, *imp-one-C* and *imp-one-top* are based on proofs found in [3] (see Section 1: (4), (6), (7) and (12)).

lemma *swap*:
  assumes $a \in A$ $b \in A$ $c \in A$
  shows $a \to^A (b \to^A c) = b \to^A (a \to^A c)$
proof −
  have $a \to^A (b \to^A c) = (a *^A b) \to^A c$
    using *assms residuation* **by** *auto*
  also
  have $\ldots = (b *^A a) \to^A c$
    using *assms mult-comm* **by** *auto*
  also
  have $\ldots = b \to^A (a \to^A c)$
    using *assms residuation* **by** *auto*
  finally
  show *?thesis*
    by *auto*
qed

lemma *imp-A*:
  assumes $a \in A$ $b \in A$
  shows $a \to^A (b \to^A a) = 1^A$
proof −
  have $a \to^A (b \to^A a) = b \to^A (a \to^A a)$
    using *assms swap* **by** *blast*
  then
  show *?thesis*
    using *assms imp-one-top* **by** *simp*
qed

2.3 Multiplication monotonicity

lemma *mult-mono*:
  assumes $a \in A$ $b \in A$ $c \in A$
  shows $(a \to^A b) \to^A ((a *^A c) \to^A (b *^A c)) = 1^A$
proof −
  have $(a \to^A b) \to^A ((a *^A c) \to^A (b *^A c)) =$
      $(a \to^A b) \to^A (a \to^A (c \to^A (b *^A c)))$
    using *assms mult-closed residuation* **by** *auto*
  also
  have $\ldots = ((a \to^A b) *^A a) \to^A (c \to^A (b *^A c))$
    using *assms imp-closed mult-closed residuation* **by** *metis*
  also
  have $\ldots = ((b \to^A a) *^A b) \to^A (c \to^A (b *^A c))$
    using *assms divisibility imp-closed mult-comm* **by** *simp*
  also
  have $\ldots = (b \to^A a) \to^A (b \to^A (c \to^A (b *^A c)))$
    using *assms imp-closed mult-closed residuation* **by** *metis*
  also
  have $\ldots = (b \to^A a) \to^A ((b *^A c) \to^A (b *^A c))$

using *assms*(*2*,*3*) *mult-closed residuation* **by** *simp*
also
have ... = 1^A
using *assms imp-closed imp-one-top mult-closed* **by** *simp*
finally
show *?thesis*
by *auto*
qed

2.4 Implication monotonicity and anti-monotonicity

lemma *imp-mono*:
assumes $a \in A$ $b \in A$ $c \in A$
shows $(a \to^A b) \to^A ((c \to^A a) \to^A (c \to^A b)) = 1^A$
proof −
have $(a \to^A b) \to^A ((c \to^A a) \to^A (c \to^A b)) =$
$(a \to^A b) \to^A (((c \to^A a) *^A c) \to^A b)$
using *assms imp-closed residuation* **by** *simp*
also
have ... $= (a \to^A b) \to^A (((a \to^A c) *^A a) \to^A b)$
using *assms divisibility imp-closed mult-comm* **by** *simp*
also
have ... $= (a \to^A b) \to^A ((a \to^A c) \to^A (a \to^A b))$
using *assms imp-closed residuation* **by** *simp*
also
have ... = 1^A
using *assms imp-A imp-closed* **by** *simp*
finally
show *?thesis*
by *auto*
qed

lemma *imp-anti-mono*:
assumes $a \in A$ $b \in A$ $c \in A$
shows $(a \to^A b) \to^A ((b \to^A c) \to^A (a \to^A c)) = 1^A$
using *assms imp-closed imp-mono swap* **by** *metis*

2.5 $(\leq^A)$ defines a partial order over A

lemma *ord-reflex*:
assumes $a \in A$
shows $a \leq^A a$
using *assms hoop-order-def* **by** *simp*

lemma *ord-trans*:
assumes $a \in A$ $b \in A$ $c \in A$ $a \leq^A b$ $b \leq^A c$
shows $a \leq^A c$
proof −
have $a \to^A c = 1^A \to^A (1^A \to^A (a \to^A c))$
using *assms*(*1*,*3*) *imp-closed imp-one-C* **by** *simp*

also
have ... = $(a \rightarrow^A b) \rightarrow^A ((b \rightarrow^A c) \rightarrow^A (a \rightarrow^A c))$
using *assms(4,5) hoop-order-def* **by** *simp*
also
have ... = 1^A
using *assms(1−3) imp-anti-mono* **by** *simp*
finally
show *?thesis*
using *hoop-order-def* **by** *auto*
qed

lemma *ord-antisymm*:
assumes $a \in A$ $b \in A$ $a \le^A b$ $b \le^A a$
shows $a = b$
proof −
have $a = a *^A (a \rightarrow^A b)$
using *assms(1,3) hoop-order-def* **by** *simp*
also
have ... = $b *^A (b \rightarrow^A a)$
using *assms(1,2) divisibility* **by** *simp*
also
have ... = b
using *assms(2,4) hoop-order-def* **by** *simp*
finally
show *?thesis*
by *auto*
qed

lemma *ord-antisymm-equiv*:
assumes $a \in A$ $b \in A$ $a \rightarrow^A b = 1^A$ $b \rightarrow^A a = 1^A$
shows $a = b$
using *assms hoop-order-def ord-antisymm* **by** *auto*

lemma *ord-top*:
assumes $a \in A$
shows $a \le^A 1^A$
using *assms hoop-order-def imp-one-top* **by** *simp*

sublocale *top-poset-on* A $(\le^A)$ $(<^A)$ 1^A
proof
show $A \neq \emptyset$
using *one-closed* **by** *blast*
next
show *reflp-on* A $(\le^A)$
using *ord-reflex reflp-onI* **by** *blast*
next
show *antisymp-on* A $(\le^A)$
using *antisymp-onI ord-antisymm* **by** *blast*
next

show *transp-on* $A\ (\leq^A)$
using *ord-trans transp-onI* **by** *blast*
next
show $x <^A y = (x \leq^A y \wedge x \neq y)$ **if** $x \in A$ $y \in A$ **for** x y
using *hoop-order-strict-def* **by** *blast*
next
show $1^A \in A$
by *simp*
next
show $x \leq^A 1^A$ **if** $x \in A$ **for** x
using *ord-top that* **by** *simp*
qed

2.6 Order properties

lemma *ord-mult-mono-A*:
assumes $a \in A$ $b \in A$ $c \in A$
shows $(a \rightarrow^A b) \leq^A ((a *^A c) \rightarrow^A (b *^A c))$
using *assms hoop-order-def mult-mono* **by** *simp*

lemma *ord-mult-mono-B*:
assumes $a \in A$ $b \in A$ $c \in A$ $a \leq^A b$
shows $(a *^A c) \leq^A (b *^A c)$
using *assms hoop-order-def imp-one-C swap mult-closed mult-mono top-closed*
by *metis*

lemma *ord-residuation*:
assumes $a \in A$ $b \in A$ $c \in A$
shows $(a *^A b) \leq^A c \longleftrightarrow a \leq^A (b \rightarrow^A c)$
using *assms hoop-order-def residuation* **by** *simp*

lemma *ord-imp-mono-A*:
assumes $a \in A$ $b \in A$ $c \in A$
shows $(a \rightarrow^A b) \leq^A ((c \rightarrow^A a) \rightarrow^A (c \rightarrow^A b))$
using *assms hoop-order-def imp-mono* **by** *simp*

lemma *ord-imp-mono-B*:
assumes $a \in A$ $b \in A$ $c \in A$ $a \leq^A b$
shows $(c \rightarrow^A a) \leq^A (c \rightarrow^A b)$
using *assms imp-closed ord-trans ord-reflex ord-residuation mult-closed*
by *metis*

lemma *ord-imp-anti-mono-A*:
assumes $a \in A$ $b \in A$ $c \in A$
shows $(a \rightarrow^A b) \leq^A ((b \rightarrow^A c) \rightarrow^A (a \rightarrow^A c))$
using *assms hoop-order-def imp-anti-mono* **by** *simp*

lemma *ord-imp-anti-mono-B*:
assumes $a \in A$ $b \in A$ $c \in A$ $a \leq^A b$

shows $(b \to^A c) \leq^A (a \to^A c)$
using *assms hoop-order-def imp-one-C swap ord-imp-mono-A top-closed*
by *metis*

lemma *ord-A*:
assumes $a \in A$ $b \in A$
shows $b \leq^A (a \to^A b)$
using *assms hoop-order-def imp-A* **by** *simp*

lemma *ord-B*:
assumes $a \in A$ $b \in A$
shows $b \leq^A ((a \to^A b) \to^A b)$
using *assms imp-closed ord-A* **by** *simp*

lemma *ord-C*:
assumes $a \in A$ $b \in A$
shows $a \leq^A ((a \to^A b) \to^A b)$
using *assms imp-one-C one-closed ord-imp-anti-mono-A* **by** *metis*

lemma *ord-D*:
assumes $a \in A$ $b \in A$ $a <^A b$
shows $b \to^A a \neq 1^A$
using *assms hoop-order-def hoop-order-strict-def ord-antisymm* **by** *auto*

2.7 Additional multiplication properties

lemma *mult-lesseq-inf*:
assumes $a \in A$ $b \in A$
shows $(a *^A b) \leq^A (a \wedge^A b)$
proof −
have $b \leq^A (a \to^A b)$
using *assms ord-A* **by** *simp*
then
have $(a *^A b) \leq^A (a *^A (a \to^A b))$
using *assms imp-closed ord-mult-mono-B mult-comm* **by** *metis*
then
show *?thesis*
using *hoop-inf-def* **by** *metis*
qed

lemma *mult-A*:
assumes $a \in A$ $b \in A$
shows $(a *^A b) \leq^A a$
using *assms ord-A ord-residuation* **by** *simp*

lemma *mult-B*:
assumes $a \in A$ $b \in A$
shows $(a *^A b) \leq^A b$
using *assms mult-A mult-comm* **by** *metis*

lemma *mult-C*:
  assumes $a \in A-\{1^A\}$ $b \in A-\{1^A\}$
  shows $a *^A b \in A-\{1^A\}$
  using *assms ord-antisymm ord-top mult-A mult-closed* **by** *force*

2.8 Additional implication properties

lemma *imp-B*:
  assumes $a \in A$ $b \in A$
  shows $a \to^A b = ((a \to^A b) \to^A b) \to^A b$
proof −
  have $a \leq^A ((a \to^A b) \to^A b)$
    using *assms ord-C* **by** *simp*
  then
  have $(((a \to^A b) \to^A b) \to^A b) \leq^A (a \to^A b)$
    using *assms imp-closed ord-imp-anti-mono-B* **by** *simp*
  moreover
  have $(a \to^A b) \leq^A (((a \to^A b) \to^A b) \to^A b)$
    using *assms imp-closed ord-C* **by** *simp*
  ultimately
  show *?thesis*
    using *assms imp-closed ord-antisymm* **by** *simp*
qed

The following two results can be found in [2] (see Proposition 1.7 and 2.2).

lemma *imp-C*:
  assumes $a \in A$ $b \in A$
  shows $(a \to^A b) \to^A (b \to^A a) = b \to^A a$
proof −
  have $a \leq^A ((a \to^A b) \to^A a)$
    using *assms imp-closed ord-A* **by** *simp*
  then
  have $(((a \to^A b) \to^A a) \to^A b) \leq^A (a \to^A b)$
    using *assms imp-closed ord-imp-anti-mono-B* **by** *simp*
  moreover
  have $(a \to^A b) \leq^A (((a \to^A b) \to^A a) \to^A a)$
    using *assms imp-closed ord-C* **by** *simp*
  ultimately
  have $(((a \to^A b) \to^A a) \to^A b) \leq^A (((a \to^A b) \to^A a) \to^A a)$
    using *assms imp-closed ord-trans* **by** *meson*
  then
  have $((((a \to^A b) \to^A a) \to^A b) *^A ((a \to^A b) \to^A a)) \leq^A a$
    using *assms imp-closed ord-residuation* **by** *simp*
  then
  have $((b \to^A ((a \to^A b) \to^A a)) *^A b) \leq^A a$
    using *assms divisibility imp-closed mult-comm* **by** *simp*
  then
  have $(b \to^A ((a \to^A b) \to^A a)) \leq^A (b \to^A a)$

using *assms imp-closed ord-residuation* **by** *simp*
then
have $((a \to^A b) \to^A (b \to^A a)) \leq^A (b \to^A a)$
using *assms imp-closed swap* **by** *simp*
moreover
have $(b \to^A a) \leq^A ((a \to^A b) \to^A (b \to^A a))$
using *assms imp-closed ord-A* **by** *simp*
ultimately
show *?thesis*
using *assms imp-closed ord-antisymm* **by** *auto*
qed

lemma *imp-D*:
assumes $a \in A$ $b \in A$
shows $(((b \to^A a) \to^A a) \to^A b) \to^A (b \to^A a) = b \to^A a$
proof −
have $(((b \to^A a) \to^A a) \to^A b) \to^A (b \to^A a) =$
$(((b \to^A a) \to^A a) \to^A b) \to^A (((b \to^A a) \to^A a) \to^A a)$
using *assms imp-B* **by** *simp*
also
have $\ldots = ((((b \to^A a) \to^A a) \to^A b) *^A ((b \to^A a) \to^A a)) \to^A a$
using *assms imp-closed residuation* **by** *simp*
also
have $\ldots = ((b \to^A ((b \to^A a) \to^A a)) *^A b) \to^A a$
using *assms divisibility imp-closed mult-comm* **by** *simp*
also
have $\ldots = (1^A *^A b) \to^A a$
using *assms hoop-order-def ord-C* **by** *simp*
also
have $\ldots = b \to^A a$
using *assms(2) mult-neutr-2* **by** *simp*
finally
show *?thesis*
by *auto*
qed

2.9 $(\wedge^A)$ defines a semilattice over A

lemma *inf-closed*:
assumes $a \in A$ $b \in A$
shows $a \wedge^A b \in A$
using *assms hoop-inf-def imp-closed mult-closed* **by** *simp*

lemma *inf-comm*:
assumes $a \in A$ $b \in A$
shows $a \wedge^A b = b \wedge^A a$
using *assms divisibility hoop-inf-def* **by** *simp*

lemma *inf-A*:

assumes $a \in A$ $b \in A$
shows $(a \wedge^A b) \leq^A a$
proof −
have $(a \wedge^A b) \rightarrow^A a = (a *^A (a \rightarrow^A b)) \rightarrow^A a$
using *hoop-inf-def* **by** *simp*
also
have $\ldots = (a \rightarrow^A b) \rightarrow^A (a \rightarrow^A a)$
using *assms mult-comm imp-closed residuation* **by** *metis*
finally
show *?thesis*
using *assms hoop-order-def imp-closed imp-one-top* **by** *simp*
qed

lemma *inf-B*:
assumes $a \in A$ $b \in A$
shows $(a \wedge^A b) \leq^A b$
using *assms inf-comm inf-A* **by** *metis*

lemma *inf-C*:
assumes $a \in A$ $b \in A$ $c \in A$ $a \leq^A b$ $a \leq^A c$
shows $a \leq^A (b \wedge^A c)$
proof −
have $(b \rightarrow^A a) \leq^A (b \rightarrow^A c)$
using *assms(1−3,5) ord-imp-mono-B* **by** *simp*
then
have $(b *^A (b \rightarrow^A a)) \leq^A (b *^A (b \rightarrow^A c))$
using *assms imp-closed ord-mult-mono-B mult-comm* **by** *metis*
moreover
have $a = b *^A (b \rightarrow^A a)$
using *assms(1−3,4) divisibility hoop-order-def mult-neutr* **by** *simp*
ultimately
show *?thesis*
using *hoop-inf-def* **by** *auto*
qed

lemma *inf-order*:
assumes $a \in A$ $b \in A$
shows $a \leq^A b \longleftrightarrow (a \wedge^A b = a)$
using *assms hoop-inf-def hoop-order-def inf-B mult-neutr* **by** *metis*

2.10 Properties of $(\vee^{*A})$

lemma *pseudo-sup-closed*:
assumes $a \in A$ $b \in A$
shows $a \vee^{*A} b \in A$
using *assms hoop-pseudo-sup-def imp-closed inf-closed* **by** *simp*

lemma *pseudo-sup-comm*:
assumes $a \in A$ $b \in A$

shows $a \vee^{*A} b = b \vee^{*A} a$
using *assms hoop-pseudo-sup-def imp-closed inf-comm* **by** *auto*

lemma *pseudo-sup-A*:
assumes $a \in A$ $b \in A$
shows $a \leq^A (a \vee^{*A} b)$
using *assms hoop-pseudo-sup-def imp-closed inf-C ord-B ord-C* **by** *simp*

lemma *pseudo-sup-B*:
assumes $a \in A$ $b \in A$
shows $b \leq^A (a \vee^{*A} b)$
using *assms pseudo-sup-A pseudo-sup-comm* **by** *metis*

lemma *pseudo-sup-order*:
assumes $a \in A$ $b \in A$
shows $a \leq^A b \longleftrightarrow a \vee^{*A} b = b$
proof
assume $a \leq^A b$
then
have $a \vee^{*A} b = b \wedge^A ((b \rightarrow^A a) \rightarrow^A a)$
using *assms(2) hoop-order-def hoop-pseudo-sup-def imp-one-C* **by** *simp*
also
have $\ldots = b$
using *assms imp-closed inf-order ord-C* **by** *meson*
finally
show $a \vee^{*A} b = b$
by *auto*
next
assume $a \vee^{*A} b = b$
then
show $a \leq^A b$
using *assms pseudo-sup-A* **by** *metis*
qed

end

end

3 Ordinal sums

We define *tower of hoops*, a family of almost disjoint hoops indexed by a total order. This is based on the definition of *bounded tower of irreducible hoops* in [5] (see paragraph after Lemma 3.3). Parting from a tower of hoops we can define a hoop known as *ordinal sum*. Ordinal sums are a fundamental tool in the study of totally ordered hoops.

theory *Ordinal-Sums*
imports *Hoops*
begin

3.1 Tower of hoops

locale *tower-of-hoops* =
  fixes *index-set* :: $'b$ *set* (I)
  fixes *index-lesseq* :: $'b \Rightarrow 'b \Rightarrow$ *bool* (**infix** $\leq^I$ *60*)
  fixes *index-less* :: $'b \Rightarrow 'b \Rightarrow$ *bool* (**infix** $<^I$ *60*)
  fixes *universes* :: $'b \Rightarrow ('a\ set)$ (UNI)
  fixes *multiplications* :: $'b \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ (MUL)
  fixes *implications* :: $'b \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ (IMP)
  fixes *sum-one* :: $'a$ (1^S)
  assumes *index-set-total-order*: *total-poset-on* I $(\leq^I)$ $(<^I)$
  and *almost-disjoint*: $i \in I \Longrightarrow j \in I \Longrightarrow i \neq j \Longrightarrow UNI\ i \cap UNI\ j = \{1^S\}$
  and *family-of-hoops*: $i \in I \Longrightarrow$ *hoop* $(UNI\ i)$ $(MUL\ i)$ $(IMP\ i)$ 1^S
begin

sublocale *total-poset-on* I $(\leq^I)$ $(<^I)$
  using *index-set-total-order* **by** *simp*

abbreviation (*uni-i*)
  uni-i :: $['b] \Rightarrow ('a\ set)$ $((\mathbb{A}(\_))$ $[61]$ *60*)
  where $\mathbb{A}_i \equiv UNI\ i$

abbreviation (*mult-i*)
  mult-i :: $['b] \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ $((*(^{-}))$ $[61]$ *60*)
  where $*^i \equiv MUL\ i$

abbreviation (*imp-i*)
  imp-i :: $['b] \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ $((\rightarrow(^{-}))$ $[61]$ *60*)
  where $\rightarrow^i \equiv IMP\ i$

abbreviation (*mult-i-xy*)
  mult-i-xy :: $['a, 'b, 'a] \Rightarrow 'a$ $(((\text{-})/\ *(^{-})\ /\ (\text{-}))$ $[61, 50, 61]$ *60*)
  where $x *^i y \equiv MUL\ i\ x\ y$

abbreviation (*imp-i-xy*)
  imp-i-xy :: $['a, 'b, 'a] \Rightarrow 'a$ $(((\text{-})/\ \rightarrow(^{-})\ /\ (\text{-}))$ $[61, 50, 61]$ *60*)
  where $x \rightarrow^i y \equiv IMP\ i\ x\ y$

3.2 Ordinal sum universe

definition *sum-univ* :: $'a$ *set* (S)
  where $S = \{x.\ \exists\ i \in I.\ x \in \mathbb{A}_i\}$

lemma *sum-one-closed* [*simp*]: $1^S \in S$
  using *family-of-hoops hoop.one-closed not-empty sum-univ-def* **by** *fastforce*

lemma *sum-subsets*:
  assumes $i \in I$
  shows $\mathbb{A}_i \subseteq S$
  using *sum-univ-def assms* **by** *blast*

3.3 Floor function: definition and properties

lemma *floor-unique*:
  assumes $a \in S - \{1^S\}$
  shows $\exists!\ i.\ i \in I \wedge a \in \mathbb{A}_i$
  using *assms sum-univ-def almost-disjoint* **by** *blast*

function *floor* :: $'a \Rightarrow 'b$ **where**
  floor $x = (THE\ i.\ i \in I \wedge x \in \mathbb{A}_i)$ **if** $x \in S - \{1^S\}$
| *floor* $x =$ *undefined* **if** $x = 1^S \vee x \notin S$
  by *auto*
termination by *lexicographic-order*

abbreviation (*uni-floor*)
  uni-floor :: $['a] \Rightarrow ('a\ set)\ ((\mathbb{A}_{floor}\ (\_))\ [61]\ 60)$
  where $\mathbb{A}_{floor\ x} \equiv UNI\ (floor\ x)$

abbreviation (*mult-floor*)
  mult-floor :: $['a] \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)\ \ ((*^{floor}\ (\bar{\ }))\ [61]\ 60)$
  where $*^{floor\ a} \equiv MUL\ (floor\ a)$

abbreviation (*imp-floor*)
  imp-floor :: $['a] \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)\ \ ((\rightarrow^{floor}\ (\bar{\ }))\ [61]\ 60)$
  where $\rightarrow^{floor\ a} \equiv IMP\ (floor\ a)$

abbreviation (*mult-floor-xy*)
  mult-floor-xy :: $['a, 'a, 'a] \Rightarrow 'a\ \ (((\text{-})/\ *^{floor}\ (\bar{\ })\ /\ (\text{-}))\ [61, 50, 61]\ 60)$
  where $x *^{floor\ y} z \equiv MUL\ (floor\ y)\ x\ z$

abbreviation (*imp-floor-xy*)
  imp-floor-xy :: $['a, 'a, 'a] \Rightarrow 'a\ \ (((\text{-})/\ \rightarrow^{floor}\ (\bar{\ })\ /\ (\text{-}))\ [61, 50, 61]\ 60)$
  where $x \rightarrow^{floor\ y} z \equiv IMP\ (floor\ y)\ x\ z$

lemma *floor-prop*:
  assumes $a \in S - \{1^S\}$
  shows *floor* $a \in I \wedge a \in \mathbb{A}_{floor\ a}$
proof −
  have *floor* $a = (THE\ i.\ i \in I \wedge a \in \mathbb{A}_i)$
    using *assms* **by** *auto*
  then
  show *?thesis*
    using *assms theI-unique floor-unique* **by** (*metis* (*mono-tags*, *lifting*))
qed

lemma *floor-one-closed*:
  assumes $i \in I$
  shows $1^S \in \mathbb{A}_i$
  using *assms floor-prop family-of-hoops hoop.one-closed* **by** *metis*

lemma *floor-mult-closed*:

assumes $i \in I$ $a \in \mathbb{A}_i$ $b \in \mathbb{A}_i$
shows $a *^i b \in \mathbb{A}_i$
using *assms family-of-hoops hoop.mult-closed* **by** *meson*

lemma *floor-imp-closed*:
assumes $i \in I$ $a \in \mathbb{A}_i$ $b \in \mathbb{A}_i$
shows $a \to^i b \in \mathbb{A}_i$
using *assms family-of-hoops hoop.imp-closed* **by** *meson*

3.4 Ordinal sum multiplication and implication

function *sum-mult* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infix** $*^S$ *60*) **where**
$x *^S y = x *^{floor\ x} y$ **if** $x \in S-\{1^S\}$ $y \in S-\{1^S\}$ *floor* x = *floor* y
| $x *^S y = x$ **if** $x \in S-\{1^S\}$ $y \in S-\{1^S\}$ *floor* $x <^I$ *floor* y
| $x *^S y = y$ **if** $x \in S-\{1^S\}$ $y \in S-\{1^S\}$ *floor* $y <^I$ *floor* x
| $x *^S y = y$ **if** $x = 1^S$ $y \in S-\{1^S\}$
| $x *^S y = x$ **if** $x \in S-\{1^S\}$ $y = 1^S$
| $x *^S y = 1^S$ **if** $x = 1^S$ $y = 1^S$
| $x *^S y =$ *undefined* **if** $x \notin S \lor y \notin S$
apply *auto*
using *floor.cases floor.simps(1) floor-prop trichotomy* **apply** (*smt* (*verit*))
using *floor-prop strict-iff-order* **apply** *force*
using *floor-prop strict-iff-order* **apply** *force*
using *floor-prop trichotomy* **by** *auto*
termination by *lexicographic-order*

function *sum-imp* :: $'a \Rightarrow 'a \Rightarrow 'a$ (**infix** $\to^S$ *60*) **where**
$x \to^S y = x \to^{floor\ x} y$ **if** $x \in S-\{1^S\}$ $y \in S-\{1^S\}$ *floor* x = *floor* y
| $x \to^S y = 1^S$ **if** $x \in S-\{1^S\}$ $y \in S-\{1^S\}$ *floor* $x <^I$ *floor* y
| $x \to^S y = y$ **if** $x \in S-\{1^S\}$ $y \in S-\{1^S\}$ *floor* $y <^I$ *floor* x
| $x \to^S y = y$ **if** $x = 1^S$ $y \in S-\{1^S\}$
| $x \to^S y = 1^S$ **if** $x \in S-\{1^S\}$ $y = 1^S$
| $x \to^S y = 1^S$ **if** $x = 1^S$ $y = 1^S$
| $x \to^S y =$ *undefined* **if** $x \notin S \lor y \notin S$
apply *auto*
using *floor.cases floor.simps(1) floor-prop trichotomy* **apply** (*smt* (*verit*))
using *floor-prop strict-iff-order* **apply** *force*
using *floor-prop strict-iff-order* **apply** *force*
using *floor-prop trichotomy* **by** *auto*
termination by *lexicographic-order*

3.4.1 Some multiplication properties

lemma *sum-mult-not-one-aux*:
assumes $a \in S-\{1^S\}$ $b \in \mathbb{A}_{floor\ a}$
shows $a *^S b \in (\mathbb{A}_{floor\ a})-\{1^S\}$
proof −
consider (*1*) $b \in S-\{1^S\}$
| (*2*) $b = 1^S$
using *sum-subsets assms floor-prop* **by** *blast*

then
show *?thesis*
proof(*cases*)
case *1*
then
have *same-floor*: *floor* a = *floor* b
using *assms floor-prop floor-unique* **by** *metis*
moreover
have $a *^S b = a *^{floor\ a} b$
using *1 assms*(*1*) *same-floor* **by** *simp*
moreover
have $a \in (\mathbb{A}_{floor\ a}){-}\{1^S\} \wedge b \in (\mathbb{A}_{floor\ a}){-}\{1^S\}$
using *1 assms floor-prop* **by** *simp*
ultimately
show *?thesis*
using *assms*(*1*) *family-of-hoops floor-prop hoop.mult-C* **by** *metis*
next
case *2*
then
show *?thesis*
using *assms*(*1*) *floor-prop* **by** *auto*
qed
qed

corollary *sum-mult-not-one*:
assumes $a \in S{-}\{1^S\}$ $b \in \mathbb{A}_{floor\ a}$
shows $a *^S b \in S{-}\{1^S\} \wedge$ *floor* $(a *^S b)$ = *floor* a
proof −
have $a *^S b \in (\mathbb{A}_{floor\ a}){-}\{1^S\}$
using *sum-mult-not-one-aux assms* **by** *meson*
then
have $a *^S b \in S{-}\{1^S\} \wedge a *^S b \in \mathbb{A}_{floor\ a}$
using *sum-subsets assms*(*1*) *floor-prop* **by** *fastforce*
then
show *?thesis*
using *assms*(*1*) *floor-prop floor-unique* **by** *metis*
qed

lemma *sum-mult-A*:
assumes $a \in S{-}\{1^S\}$ $b \in \mathbb{A}_{floor\ a}$
shows $a *^S b = a *^{floor\ a} b \wedge b *^S a = b *^{floor\ a} a$
proof −
consider (*1*) $b \in S{-}\{1^S\}$
| (*2*) $b = 1^S$
using *sum-subsets assms floor-prop* **by** *blast*
then
show *?thesis*
proof(*cases*)
case *1*

then
have *floor* a = *floor* b
using *assms floor.cases floor-prop floor-unique* **by** *metis*
then
show *?thesis*
using *1 assms* **by** *auto*
next
case *2*
then
show *?thesis*
using *assms*(*1*) *family-of-hoops floor-prop hoop.mult-neutr hoop.mult-neutr-2*
by *fastforce*
qed
qed

3.4.2 Some implication properties

lemma *sum-imp-floor*:
assumes $a \in S - \{1^S\}$ $b \in S - \{1^S\}$ *floor* a = *floor* b $a \to^S b \in S - \{1^S\}$
shows *floor* $(a \to^S b)$ = *floor* a
proof −
have $a \to^S b \in \mathbb{A}_{floor\ a}$
using *sum-imp.simps*(*1*) *assms*(*1*−*3*) *floor-imp-closed floor-prop*
by *metis*
then
show *?thesis*
using *assms*(*1*,*4*) *floor-prop floor-unique* **by** *blast*
qed

lemma *sum-imp-A*:
assumes $a \in S - \{1^S\}$ $b \in \mathbb{A}_{floor\ a}$
shows $a \to^S b = a \to^{floor\ a} b$
proof −
consider (*1*) $b \in S - \{1^S\}$
| (*2*) $b = 1^S$
using *sum-subsets assms floor-prop* **by** *blast*
then
show *?thesis*
proof(*cases*)
case *1*
then
show *?thesis*
using *sum-imp.simps*(*1*) *assms floor-prop floor-unique* **by** *metis*
next
case *2*
then
show *?thesis*
using *sum-imp.simps*(*5*) *assms*(*1*) *family-of-hoops floor-prop*
hoop.imp-one-top

    by *metis*
  qed
qed

lemma *sum-imp-B*:
  assumes $a \in S-\{1^S\}$ $b \in \mathbb{A}_{floor\ a}$
  shows $b \to^S a = b \to^{floor\ a} a$
proof −
  consider (*1*) $b \in S-\{1^S\}$
    | (*2*) $b = 1^S$
    using *sum-subsets assms floor-prop* **by** *blast*
  then
  show *?thesis*
  proof(*cases*)
    case *1*
    then
    show *?thesis*
      using *sum-imp.simps*(*1*) *assms floor-prop floor-unique* **by** *metis*
  next
    case *2*
    then
    show *?thesis*
      using *sum-imp.simps*(*4*) *assms*(*1*) *family-of-hoops floor-prop*
          hoop.imp-one-C
      by *metis*
  qed
qed

lemma *sum-imp-floor-antisymm*:
  assumes $a \in S-\{1^S\}$ $b \in S-\{1^S\}$ *floor* $a =$ *floor* b
        $a \to^S b = 1^S$ $b \to^S a = 1^S$
  shows $a = b$
proof −
  have $a \in \mathbb{A}_{floor\ a} \wedge b \in \mathbb{A}_{floor\ a} \wedge$ *floor* $a \in I$
    using *floor-prop assms* **by** *metis*
  moreover
  have $a \to^S b = a \to^{floor\ a} b \wedge b \to^S a = b \to^{floor\ a} a$
    using *assms* **by** *auto*
  ultimately
  show *?thesis*
    using *assms*(*4*,*5*) *family-of-hoops hoop.ord-antisymm-equiv* **by** *metis*
qed

corollary *sum-imp-C*:
  assumes $a \in S-\{1^S\}$ $b \in S-\{1^S\}$ $a \neq b$ *floor* $a =$ *floor* b $a \to^S b = 1^S$
  shows $b \to^S a \neq 1^S$
  using *sum-imp-floor-antisymm assms* **by** *blast*

lemma *sum-imp-D*:

assumes $a \in S$
shows $1^S \rightarrow^S a = a$
using *sum-imp.simps(4,6) assms* **by** *blast*

lemma *sum-imp-E*:
assumes $a \in S$
shows $a \rightarrow^S 1^S = 1^S$
using *sum-imp.simps(5,6) assms* **by** *blast*

3.5 The ordinal sum of a tower of hoops is a hoop

3.5.1 S is not empty

lemma *sum-not-empty*: $S \neq \emptyset$
using *sum-one-closed* **by** *blast*

3.5.2 $(*^S)$ and $(\rightarrow^S)$ are well defined

lemma *sum-mult-closed-one*:
assumes $a \in S$ $b \in S$ $a = 1^S \vee b = 1^S$
shows $a *^S b \in S$
using *sum-mult.simps(4−6) assms floor.cases* **by** *metis*

lemma *sum-mult-closed-not-one*:
assumes $a \in S - \{1^S\}$ $b \in S - \{1^S\}$
shows $a *^S b \in S - \{1^S\}$
proof −
from *assms*
consider *(1) floor a = floor b*
| *(2) floor a* $<^I$ *floor b* $\vee$ *floor b* $<^I$ *floor a*
using *trichotomy floor-prop* **by** *blast*
then
show *?thesis*
proof(*cases*)
case *1*
then
show *?thesis*
using *sum-mult-not-one assms floor-prop* **by** *metis*
next
case *2*
then
show *?thesis*
using *assms* **by** *auto*
qed
qed

lemma *sum-mult-closed*:
assumes $a \in S$ $b \in S$
shows $a *^S b \in S$
using *sum-mult-closed-not-one sum-mult-closed-one assms* **by** *auto*

lemma *sum-imp-closed-one*:
  assumes $a \in S$ $b \in S$ $a = 1^S \lor b = 1^S$
  shows $a \to^S b \in S$
  using *sum-imp.simps(4−6) assms floor.cases* **by** *metis*

lemma *sum-imp-closed-not-one*:
  assumes $a \in S-\{1^S\}$ $b \in S-\{1^S\}$
  shows $a \to^S b \in S$
proof −
  from *assms*
  consider (1) *floor* a = *floor* b
    | (2) *floor* $a <^I$ *floor* b ∨ *floor* $b <^I$ *floor* a
    using *trichotomy floor-prop* **by** *blast*
  then
  show $a \to^S b \in S$
  proof(*cases*)
    case *1*
    then
    have $a \to^S b = a \to^{floor\ a} b$
      using *assms* **by** *auto*
    moreover
    have $a \to^{floor\ a} b \in \mathbb{A}_{floor\ a}$
      using *1 assms floor-imp-closed floor-prop* **by** *metis*
    ultimately
    show *?thesis*
      using *sum-subsets assms(1) floor-prop* **by** *auto*
  next
    case *2*
    then
    show *?thesis*
      using *assms* **by** *auto*
  qed
qed

lemma *sum-imp-closed*:
  assumes $a \in S$ $b \in S$
  shows $a \to^S b \in S$
  using *sum-imp-closed-one sum-imp-closed-not-one assms* **by** *auto*

3.5.3 Neutrality of 1^S

lemma *sum-mult-neutr*:
  assumes $a \in S$
  shows $a *^S 1^S = a \land 1^S *^S a = a$
  using *assms sum-mult.simps(4−6)* **by** *blast*

3.5.4 Commutativity of $(*^S)$

Now we prove $x *^S y = y *^S x$ by showing that it holds when one of the variables is equal to 1^S. Then we consider when none of them is 1^S.

lemma *sum-mult-comm-one*:
  assumes $a \in S$ $b \in S$ $a = 1^S \vee b = 1^S$
  shows $a *^S b = b *^S a$
  using *sum-mult-neutr assms* **by** *auto*

lemma *sum-mult-comm-not-one*:
  assumes $a \in S - \{1^S\}$ $b \in S - \{1^S\}$
  shows $a *^S b = b *^S a$
proof −
  from *assms*
  consider (*1*) *floor a* = *floor b*
    | (*2*) *floor a* $<^I$ *floor b* ∨ *floor b* $<^I$ *floor a*
    using *trichotomy floor-prop* **by** *blast*
  then
  show *?thesis*
  proof(*cases*)
    case *1*
    then
    have *same-floor*: $b \in \mathbb{A}_{floor\ a}$
      using *assms*(*2*) *floor-prop* **by** *simp*
    then
    have $a *^S b = a *^{floor\ a} b$
      using *sum-mult-A assms*(*1*) **by** *blast*
    also
    have $\ldots = b *^{floor\ a} a$
      using *assms*(*1*) *family-of-hoops floor-prop hoop.mult-comm same-floor*
      by *meson*
    also
    have $\ldots = b *^S a$
      using *sum-mult-A assms*(*1*) *same-floor* **by** *simp*
    finally
    show *?thesis*
      by *auto*
  next
    case *2*
    then
    show *?thesis*
      using *assms* **by** *auto*
  qed
qed

lemma *sum-mult-comm*:
  assumes $a \in S$ $b \in S$
  shows $a *^S b = b *^S a$
  using *assms sum-mult-comm-one sum-mult-comm-not-one* **by** *auto*

3.5.5 Associativity of $(*^S)$

Next we prove $x *^S (y *^S z) = (x *^S y) *^S z$.

lemma *sum-mult-assoc-one*:
 assumes $a \in S$ $b \in S$ $c \in S$ $a = 1^S \vee b = 1^S \vee c = 1^S$
 shows $a *^S (b *^S c) = (a *^S b) *^S c$
 using *sum-mult-neutr assms sum-mult-closed* **by** *metis*

lemma *sum-mult-assoc-not-one*:
 assumes $a \in S-\{1^S\}$ $b \in S-\{1^S\}$ $c \in S-\{1^S\}$
 shows $a *^S (b *^S c) = (a *^S b) *^S c$
proof −
 from *assms*
 consider (*1*) *floor* a = *floor* b *floor* b = *floor* c
  | (*2*) *floor* a = *floor* b *floor* $b <^I$ *floor* c
  | (*3*) *floor* a = *floor* b *floor* $c <^I$ *floor* b
  | (*4*) *floor* $a <^I$ *floor* b *floor* b = *floor* c
  | (*5*) *floor* $a <^I$ *floor* b *floor* $b <^I$ *floor* c
  | (*6*) *floor* $a <^I$ *floor* b *floor* $c <^I$ *floor* b
  | (*7*) *floor* $b <^I$ *floor* a *floor* b = *floor* c
  | (*8*) *floor* $b <^I$ *floor* a *floor* $b <^I$ *floor* c
  | (*9*) *floor* $b <^I$ *floor* a *floor* $c <^I$ *floor* b
  using *trichotomy floor-prop* **by** *meson*
 then
 show *?thesis*
 proof(*cases*)
  case *1*
  then
  have $a *^S (b *^S c) = a *^{floor\ a} (b *^{floor\ a} c)$
   using *sum-mult-A assms floor-mult-closed floor-prop* **by** *metis*
  also
  have $\ldots = (a *^{floor\ a} b) *^{floor\ a} c$
   using *1 assms family-of-hoops floor-prop hoop.mult-assoc* **by** *metis*
  also
  have $\ldots = (a *^{floor\ b} b) *^{floor\ b} c$
   using *1* **by** *simp*
  also
  have $\ldots = (a *^S b) *^S c$
   using *1 sum-mult-A assms floor-mult-closed floor-prop* **by** *metis*
  finally
  show *?thesis*
   by *auto*
 next
  case *2*
  then
  show *?thesis*
   using *sum-mult.simps(2,3) sum-mult-not-one assms floor-prop* **by** *metis*
 next
  case *3*

then
show *?thesis*
using *sum-mult.simps(3) sum-mult-not-one assms floor-prop* **by** *metis*
next
case *4*
then
show *?thesis*
using *sum-mult.simps(2) sum-mult-not-one assms floor-prop* **by** *metis*
next
case *5*
then
show *?thesis*
using *sum-mult.simps(2) assms floor-prop strict-trans* **by** *metis*
next
case *6*
then
show *?thesis*
using *sum-mult.simps(2,3) assms* **by** *metis*
next
case *7*
then
show *?thesis*
using *sum-mult.simps(3) sum-mult-not-one assms floor-prop* **by** *metis*
next
case *8*
then
show *?thesis*
using *sum-mult.simps(2,3) assms* **by** *metis*
next
case *9*
then
show *?thesis*
using *sum-mult.simps(3) assms floor-prop strict-trans* **by** *metis*
qed
qed

lemma *sum-mult-assoc*:
assumes $a \in S$ $b \in S$ $c \in S$
shows $a *^S (b *^S c) = (a *^S b) *^S c$
using *assms sum-mult-assoc-one sum-mult-assoc-not-one* **by** *blast*

3.5.6 Reflexivity of $(\rightarrow^S)$

lemma *sum-imp-reflex*:
assumes $a \in S$
shows $a \rightarrow^S a = 1^S$
proof −
consider (1) $a \in S-\{1^S\}$
| (2) $a = 1^S$

using *assms* **by** *blast*
then
show *?thesis*
proof(*cases*)
case *1*
then
have $a \to^S a = a \to^{floor\ a} a$
by *simp*
then
show *?thesis*
using *1 family-of-hoops floor-prop hoop.imp-reflex* **by** *metis*
next
case *2*
then
show *?thesis*
by *simp*
qed
qed

3.5.7 Divisibility

We prove $x *^S (x \to^S y) = y *^S (y \to^S x)$ using the same methods as before.

lemma *sum-divisibility-one*:
assumes $a \in S$ $b \in S$ $a = 1^S \vee b = 1^S$
shows $a *^S (a \to^S b) = b *^S (b \to^S a)$
proof −
have $x \to^S y = y \wedge y \to^S x = 1^S$ **if** $x = 1^S$ $y \in S$ **for** x y
using *sum-imp-D sum-imp-E that* **by** *simp*
then
show *?thesis*
using *assms sum-mult-neutr* **by** *metis*
qed

lemma *sum-divisibility-aux*:
assumes $a \in S - \{1^S\}$ $b \in \mathbb{A}_{floor\ a}$
shows $a *^S (a \to^S b) = a *^{floor\ a} (a \to^{floor\ a} b)$
using *sum-imp-A sum-mult-A assms floor-imp-closed floor-prop* **by** *metis*

lemma *sum-divisibility-not-one*:
assumes $a \in S - \{1^S\}$ $b \in S - \{1^S\}$
shows $a *^S (a \to^S b) = b *^S (b \to^S a)$
proof −
from *assms*
consider (*1*) *floor a* = *floor b*
| (*2*) *floor a* $<^I$ *floor b* ∨ *floor b* $<^I$ *floor a*
using *trichotomy floor-prop* **by** *blast*
then
show *?thesis*
proof(*cases*)

case *1*
then
have $a *^S (a \to^S b) = a *^{floor\ a} (a \to^{floor\ a} b)$
using *1 sum-divisibility-aux assms floor-prop* **by** *metis*
also
have $\ldots = b *^{floor\ a} (b \to^{floor\ a} a)$
using *1 assms family-of-hoops floor-prop hoop.divisibility* **by** *metis*
also
have $\ldots = b *^{floor\ b} (b \to^{floor\ b} a)$
using *1* **by** *simp*
also
have $\ldots = b *^S (b \to^S a)$
using *1 sum-divisibility-aux assms floor-prop* **by** *metis*
finally
show *?thesis*
by *auto*
next
case *2*
then
show *?thesis*
using *assms* **by** *auto*
qed
qed

lemma *sum-divisibility*:
assumes $a \in S$ $b \in S$
shows $a *^S (a \to^S b) = b *^S (b \to^S a)$
using *assms sum-divisibility-one sum-divisibility-not-one* **by** *auto*

3.5.8 Residuation

Finally we prove $(x *^S y) \to^S z = x \to^S (y \to^S z)$.

lemma *sum-residuation-one*:
assumes $a \in S$ $b \in S$ $c \in S$ $a = 1^S \vee b = 1^S \vee c = 1^S$
shows $(a *^S b) \to^S c = a \to^S (b \to^S c)$
using *sum-imp-D sum-imp-E sum-imp-closed sum-mult-closed sum-mult-neutr assms*
by *metis*

lemma *sum-residuation-not-one*:
assumes $a \in S-\{1^S\}$ $b \in S-\{1^S\}$ $c \in S-\{1^S\}$
shows $(a *^S b) \to^S c = a \to^S (b \to^S c)$
proof −
from *assms*
consider (*1*) *floor a* = *floor b* *floor b* = *floor c*
| (*2*) *floor a* = *floor b* *floor b* $<^I$ *floor c*
| (*3*) *floor a* = *floor b* *floor c* $<^I$ *floor b*
| (*4*) *floor a* $<^I$ *floor b* *floor b* = *floor c*
| (*5*) *floor a* $<^I$ *floor b* *floor b* $<^I$ *floor c*

| (*6*) *floor a* $<^I$ *floor b floor c* $<^I$ *floor b*
| (*7*) *floor b* $<^I$ *floor a floor b* = *floor c*
| (*8*) *floor b* $<^I$ *floor a floor b* $<^I$ *floor c*
| (*9*) *floor b* $<^I$ *floor a floor c* $<^I$ *floor b*
using *trichotomy floor-prop* **by** *meson*
then
show *?thesis*
proof(*cases*)
case *1*
then
have $(a *^S b) \to^S c = (a *^{floor\ a} b) \to^{floor\ a} c$
using *sum-imp-B sum-mult-A assms floor-mult-closed floor-prop* **by** *metis*
also
have $\ldots = a \to^{floor\ a} (b \to^{floor\ a} c)$
using *1 assms family-of-hoops floor-prop hoop.residuation* **by** *metis*
also
have $\ldots = a \to^{floor\ b} (b \to^{floor\ b} c)$
using *1* **by** *simp*
also
have $\ldots = a \to^S (b \to^S c)$
using *1 sum-imp-A assms floor-imp-closed floor-prop* **by** *metis*
finally
show *?thesis*
by *auto*
next
case *2*
then
show *?thesis*
using *sum-imp.simps*(*2*,*5*) *sum-mult-not-one assms floor-prop* **by** *metis*
next
case *3*
then
show *?thesis*
using *sum-imp.simps*(*3*) *sum-mult-not-one assms floor-prop* **by** *metis*
next
case *4*
then
have $(a *^S b) \to^S c = 1^S$
using *4 sum-imp.simps*(*2*) *sum-mult.simps*(*2*) *assms* **by** *metis*
moreover
have $b \to^S c = 1^S \lor (b \to^S c \in S - \{1^S\} \land floor\ (b \to^S c) = floor\ b)$
using *4*(*2*) *sum-imp-closed-not-one sum-imp-floor assms*(*2*,*3*) **by** *blast*
ultimately
show *?thesis*
using *4*(*1*) *sum-imp.simps*(*2*,*5*) *assms*(*1*) **by** *metis*
next
case *5*
then
show *?thesis*

using *sum-imp.simps(2,5) sum-mult.simps(2) assms floor-prop strict-trans*
by *metis*
next
case *6*
then
show *?thesis*
using *assms* **by** *auto*
next
case *7*
then
have $(a *^S b) \to^S c = (b \to^S c)$
using *assms(1,2)* **by** *auto*
moreover
have $b \to^S c = 1^S \lor (b \to^S c \in S - \{1^S\} \land floor\ (b \to^S c) = floor\ b)$
using *7(2) sum-imp-closed-not-one sum-imp-floor assms(2,3)* **by** *blast*
ultimately
show *?thesis*
using *7(1) sum-imp.simps(3,5) assms(1)* **by** *metis*
next
case *8*
then
show *?thesis*
using *assms* **by** *auto*
next
case *9*
then
show *?thesis*
using *sum-imp.simps(3) sum-mult.simps(3) assms floor-prop strict-trans*
by *metis*
qed
qed

lemma *sum-residuation*:
assumes $a \in S$ $b \in S$ $c \in S$
shows $(a *^S b) \to^S c = a \to^S (b \to^S c)$
using *assms sum-residuation-one sum-residuation-not-one* **by** *blast*

3.5.9 Main result

sublocale *hoop* S $(*^S)$ $(\to^S)$ 1^S
proof
show $x *^S y \in S$ **if** $x \in S$ $y \in S$ **for** x y
using *that sum-mult-closed* **by** *simp*
next
show $x \to^S y \in S$ **if** $x \in S$ $y \in S$ **for** x y
using *that sum-imp-closed* **by** *simp*
next
show $1^S \in S$
by *simp*

next
  show $x *^S\ y = y *^S\ x$ **if** $x \in S$ $y \in S$ **for** x y
    using *that sum-mult-comm* **by** *simp*
next
  show $x *^S\ (y *^S\ z) = (x *^S\ y) *^S\ z$ **if** $x \in S$ $y \in S$ $z \in S$ **for** x y z
    using *that sum-mult-assoc* **by** *simp*
next
  show $x *^S\ 1^S = x$ **if** $x \in S$ **for** x
    using *that sum-mult-neutr* **by** *simp*
next
  show $x \to^S\ x = 1^S$ **if** $x \in S$ **for** x
    using *that sum-imp-reflex* **by** *simp*
next
  show $x *^S\ (x \to^S\ y) = y *^S\ (y \to^S\ x)$ **if** $x \in S$ $y \in S$ **for** x y
    using *that sum-divisibility* **by** *simp*
next
  show $x \to^S\ (y \to^S\ z) = (x *^S\ y) \to^S\ z$ **if** $x \in S$ $y \in S$ $z \in S$ **for** x y z
    using *that sum-residuation* **by** *simp*
qed

end

end

4 Totally ordered hoops

theory *Totally-Ordered-Hoops*
  imports *Ordinal-Sums*
begin

4.1 Definitions

locale *totally-ordered-hoop* = *hoop* +
  assumes *total-order*: $x \in A \Longrightarrow y \in A \Longrightarrow x \le^A\ y \vee y \le^A\ x$
begin

function *fixed-points* :: $'a \Rightarrow 'a\ set$ (F) **where**
  $F\ a = \{b \in A - \{1^A\}.\ a \to^A\ b = b\}$ **if** $a \in A - \{1^A\}$
| $F\ a = \{1^A\}$ **if** $a = 1^A$
| $F\ a = undefined$ **if** $a \notin A$
  by *auto*
termination by *lexicographic-order*

definition *rel-F* :: $'a \Rightarrow 'a \Rightarrow bool$ (**infix** $\sim F$ *60*)
  where $x \sim F\ y \equiv \forall\ z \in A.\ (x \to^A\ z = z) \longleftrightarrow (y \to^A\ z = z)$

definition *rel-F-canonical-map* :: $'a \Rightarrow 'a\ set$ (π)
  where $\pi\ x = \{b \in A.\ x \sim F\ b\}$

end

4.2 Properties of F

context *totally-ordered-hoop*
begin

lemma *F-equiv*:
  assumes $a \in A-\{1^A\}$ $b \in A$
  shows $b \in F\ a \longleftrightarrow (b \in A \wedge b \neq 1^A \wedge a \rightarrow^A b = b)$
  using *assms* **by** *auto*

lemma *F-subset*:
  assumes $a \in A$
  shows $F\ a \subseteq A$
proof −
  have $a = 1^A \vee a \neq 1^A$
    by *auto*
  then
  show *?thesis*
    using *assms* **by** *fastforce*
qed

lemma *F-of-one*:
  assumes $a \in A$
  shows $F\ a = \{1^A\} \longleftrightarrow a = 1^A$
  using *F-equiv assms fixed-points.simps*(*2*) *top-closed* **by** *blast*

lemma *F-of-mult*:
  assumes $a \in A-\{1^A\}$ $b \in A-\{1^A\}$
  shows $F\ (a *^A b) = \{c \in A-\{1^A\}.\ (a *^A b) \rightarrow^A c = c\}$
  using *assms mult-C* **by** *auto*

lemma *F-of-imp*:
  assumes $a \in A$ $b \in A$ $a \rightarrow^A b \neq 1^A$
  shows $F\ (a \rightarrow^A b) = \{c \in A-\{1^A\}.\ (a \rightarrow^A b) \rightarrow^A c = c\}$
  using *assms imp-closed* **by** *auto*

lemma *F-bound*:
  assumes $a \in A$ $b \in A$ $a \in F\ b$
  shows $a \leq^A b$
proof −
  consider (*1*) $b \neq 1^A$
    | (*2*) $b = 1^A$
    by *auto*
  then
  show *?thesis*
  proof(*cases*)
    case *1*

then
have $b \to^A a \neq 1^A$
using *assms*(*2*,*3*) **by** *simp*
then
show *?thesis*
using *assms hoop-order-def total-order* **by** *auto*
next
case *2*
then
show *?thesis*
using *assms*(*1*) *ord-top* **by** *auto*
qed
qed

The following results can be found in Lemma 3.3 in [5].

lemma *LEMMA-3-3-1*:
assumes $a \in A-\{1^A\}$ $b \in A$ $c \in A$ $b \in F\ a$ $c \leq^A b$
shows $c \in F\ a$
proof −
from *assms*
have $(a \to^A c) \leq^A (a \to^A b)$
using *DiffD1 F-equiv ord-imp-mono-B* **by** *metis*
then
have $(a \to^A c) \leq^A b$
using *assms*(*1*,*4*,*5*) **by** *simp*
then
have $(a \to^A c) \to^A c = ((a \to^A c) *^A ((a \to^A c) \to^A b)) \to^A c$
using *assms*(*1*,*3*) *hoop-order-def imp-closed* **by** *force*
also
have $\ldots = (b *^A (b \to^A (a \to^A c))) \to^A c$
using *assms divisibility imp-closed* **by** *simp*
also
have $\ldots = (b \to^A (a \to^A c)) \to^A (b \to^A c)$
using *DiffD1 assms*(*1*−*3*) *imp-closed swap residuation* **by** *metis*
also
have $\ldots = ((a \to^A b) \to^A (a \to^A c)) \to^A (b \to^A c)$
using *assms*(*1*,*4*) **by** *simp*
also
have $\ldots = (((a \to^A b) *^A a) \to^A c) \to^A (b \to^A c)$
using *assms*(*1*,*3*,*4*) *residuation* **by** *simp*
also
have $\ldots = (((b \to^A a) *^A b) \to^A c) \to^A (b \to^A c)$
using *assms*(*1*,*2*) *divisibility imp-closed mult-comm* **by** *simp*
also
have $\ldots = (b \to^A c) \to^A (b \to^A c)$
using *F-bound assms*(*1*,*4*) *hoop-order-def* **by** *simp*
also
have $\ldots = 1^A$
using *F-bound assms hoop-order-def imp-closed* **by** *simp*

finally
have $(a \to^A c) \leq^A c$
using *hoop-order-def* **by** *simp*
moreover
have $c \leq^A (a \to^A c)$
using *assms*(*1*,*3*) *ord-A* **by** *simp*
ultimately
have $a \to^A c = c$
using *assms*(*1*,*3*) *imp-closed ord-antisymm* **by** *simp*
moreover
have $c \in A-\{1^A\}$
using *assms*(*1*,*3*−*5*) *hoop-order-def imp-one-C* **by** *auto*
ultimately
show *?thesis*
using *F-equiv assms*(*1*) **by** *blast*
qed

lemma *LEMMA-3-3-2*:
assumes $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ $F\ a = F\ b$
shows $F\ a = F\ (a *^A b)$
proof
show $F\ a \subseteq F\ (a *^A b)$
proof
fix c
assume $c \in F\ a$
then
have $(a *^A b) \to^A c = b \to^A (a \to^A c)$
using *DiffD1 F-subset assms*(*1*,*2*) *in-mono swap residuation* **by** *metis*
also
have $\ldots = b \to^A c$
using ‹$c \in F\ a$› *assms*(*1*) **by** *auto*
also
have $\ldots = c$
using ‹$c \in F\ a$› *assms*(*2*,*3*) **by** *auto*
finally
show $c \in F\ (a *^A b)$
using ‹$c \in F\ a$› *assms*(*1*,*2*) *mult-C* **by** *auto*
qed
next
show $F\ (a *^A b) \subseteq F\ a$
proof
fix c
assume $c \in F\ (a *^A b)$
then
have $(a *^A b) \leq^A a$
using *assms*(*1*,*2*) *mult-A* **by** *auto*
then
have $(a \to^A c) \leq^A ((a *^A b) \to^A c)$
using *DiffD1 F-subset* ‹$c \in F\ (a *^A b)$› *assms mult-closed*

ord-imp-anti-mono-B subsetD
by *meson*
moreover
have $(a *^A b) \to^A c = c$
using ‹$c \in F (a *^A b)$› *assms(1,2) mult-C* **by** *auto*
ultimately
have $(a \to^A c) \leq^A c$
by *simp*
moreover
have $c \leq^A (a \to^A c)$
using *DiffD1 F-subset* ‹$c \in F (a *^A b)$› *assms(1,2) insert-Diff*
insert-subset mult-closed ord-A
by *metis*
ultimately
show $c \in F a$
using ‹$c \in F (a *^A b)$› *assms(1,2) imp-closed mult-C ord-antisymm* **by** *auto*
qed
qed

lemma *LEMMA-3-3-3*:
assumes $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ $a \leq^A b$
shows $F a \subseteq F b$
proof
fix c
assume $c \in F a$
then
have $(b \to^A c) \leq^A (a \to^A c)$
using *DiffD1 F-subset assms in-mono ord-imp-anti-mono-B* **by** *meson*
moreover
have $a \to^A c = c$
using ‹$c \in F a$› *assms(1)* **by** *auto*
ultimately
have $(b \to^A c) \leq^A c$
by *simp*
moreover
have $c \leq^A (b \to^A c)$
using ‹$c \in F a$› *assms(1,2) ord-A* **by** *force*
ultimately
show $c \in F b$
using ‹$c \in F a$› *assms(1,2) imp-closed ord-antisymm* **by** *auto*
qed

lemma *LEMMA-3-3-4*:
assumes $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ $a <^A b$ $F a \neq F b$
shows $a \in F b$
proof −
from *assms*
obtain c **where** $c \in F b \wedge c \notin F a$
using *LEMMA-3-3-3 hoop-order-strict-def* **by** *auto*

then
have *witness*: $c \in A-\{1^A\} \wedge b \to^A c = c \wedge c <^A (a \to^A c)$
using *DiffD1 assms*(*1*,*2*) *hoop-order-strict-def ord-A* **by** *auto*
then
have $(a \to^A c) \to^A c \in F\ b$
using *DiffD1 F-equiv assms*(*1*,*2*) *imp-closed swap ord-D* **by** *metis*
moreover
have $a \leq^A ((a \to^A c) \to^A c)$
using *assms*(*1*) *ord-C witness* **by** *force*
ultimately
show $a \in F\ b$
using *Diff-iff LEMMA-3-3-1 assms*(*1*,*2*) *imp-closed witness* **by** *metis*
qed

lemma *LEMMA-3-3-5*:
assumes $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ $F\ a \neq F\ b$
shows $a *^A b = a \wedge^A b$
proof −
have $a <^A b \vee b <^A a$
using *DiffD1 assms hoop-order-strict-def total-order* **by** *metis*
then
have $a \in F\ b \vee b \in F\ a$
using *LEMMA-3-3-4 assms* **by** *metis*
then
have $a *^A b = (b \to^A a) *^A b \vee a *^A b = a *^A (a \to^A b)$
using *assms*(*1*,*2*) **by** *force*
then
show *?thesis*
using *assms*(*1*,*2*) *divisibility hoop-inf-def imp-closed mult-comm* **by** *auto*
qed

lemma *LEMMA-3-3-6*:
assumes $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ $a <^A b$ $F\ a = F\ b$
shows $F\ (b \to^A a) = F\ b$
proof −
have $a \notin F\ a$
using *assms*(*1*) *DiffD1 F-equiv imp-reflex* **by** *metis*
then
have $a <^A (b \to^A a)$
using *assms*(*1*,*2*,*4*) *hoop-order-strict-def ord-A* **by** *auto*
moreover
have $b *^A (b \to^A a) = a$
using *assms*(*1*−*3*) *divisibility hoop-order-def hoop-order-strict-def* **by** *simp*
moreover
have $b \leq^A (b \to^A a) \vee (b \to^A a) \leq^A b$
using *DiffD1 assms*(*1*,*2*) *imp-closed ord-reflex total-order* **by** *metis*
ultimately
have $b *^A (b \to^A a) \neq b \wedge^A (b \to^A a)$
using *assms*(*1*−*3*) *hoop-order-strict-def imp-closed inf-comm inf-order* **by** *force*

then
show $F\ (b \rightarrow^A\ a) = F\ b$
using *LEMMA-3-3-5 assms*$(1-3)$ *imp-closed ord-D* **by** *blast*
qed

4.3 Properties of $(\sim F)$

4.3.1 $(\sim F)$ is an equivalence relation

lemma *rel-F-reflex*:
assumes $a \in A$
shows $a \sim F\ a$
using *rel-F-def* **by** *auto*

lemma *rel-F-symm*:
assumes $a \in A\ \ b \in A\ \ a \sim F\ b$
shows $b \sim F\ a$
using *assms rel-F-def* **by** *auto*

lemma *rel-F-trans*:
assumes $a \in A\ \ b \in A\ \ c \in A\ \ a \sim F\ b\ \ b \sim F\ c$
shows $a \sim F\ c$
using *assms rel-F-def* **by** *auto*

4.3.2 Equivalent definition

lemma *rel-F-equiv*:
assumes $a \in A\ \ b \in A$
shows $(a \sim F\ b) = (F\ a = F\ b)$
proof
assume $a \sim F\ b$
then
consider $(1)\ a \neq 1^A\ \ b \neq 1^A$
| $(2)\ a = 1^A\ \ b = 1^A$
using *assms imp-one-C rel-F-def* **by** *fastforce*
then
show $F\ a = F\ b$
proof(*cases*)
case *1*
then
show *?thesis*
using ‹$a \sim F\ b$› *assms rel-F-def* **by** *auto*
next
case *2*
then
show *?thesis*
by *simp*
qed
next
assume $F\ a = F\ b$

then
consider (*1*) $a \neq 1^A$ $b \neq 1^A$
| (*2*) $a = 1^A$ $b = 1^A$
using *F-of-one assms* **by** *blast*
then
show $a \sim F\ b$
proof(*cases*)
case *1*
then
show *?thesis*
using ‹$F\ a = F\ b$› *assms imp-one-A imp-one-C rel-F-def* **by** *auto*
next
case *2*
then
show *?thesis*
using *rel-F-reflex* **by** *simp*
qed
qed

4.3.3 Properties of equivalence classes given by ($\sim F$)

lemma *class-one*: $\pi\ 1^A = \{1^A\}$
using *imp-one-C rel-F-canonical-map-def rel-F-def* **by** *auto*

lemma *classes-subsets*:
assumes $a \in A$
shows $\pi\ a \subseteq A$
using *rel-F-canonical-map-def* **by** *simp*

lemma *classes-not-empty*:
assumes $a \in A$
shows $a \in \pi\ a$
using *assms rel-F-canonical-map-def rel-F-reflex* **by** *simp*

corollary *class-not-one*:
assumes $a \in A - \{1^A\}$
shows $\pi\ a \neq \{1^A\}$
using *assms classes-not-empty* **by** *blast*

lemma *classes-disjoint*:
assumes $a \in A$ $b \in A$ $\pi\ a \cap \pi\ b \neq \emptyset$
shows $\pi\ a = \pi\ b$
using *assms rel-F-canonical-map-def rel-F-def rel-F-trans* **by** *force*

lemma *classes-cover*: $A = \{x.\ \exists\ y \in A.\ x \in \pi\ y\}$
using *classes-subsets classes-not-empty* **by** *auto*

lemma *classes-convex*:
assumes $a \in A$ $b \in A$ $c \in A$ $d \in A$ $b \in \pi\ a$ $c \in \pi\ a$ $b \leq^A d$ $d \leq^A c$

```
  shows d ∈ π a
proof −
  have eq-F: F a = F b ∧ F a = F c
    using assms(1,5,6) rel-F-canonical-map-def rel-F-equiv by auto
  from assms
  consider (1) c = 1^A
    | (2) c ≠ 1^A
    by auto
  then
  show ?thesis
  proof(cases)
    case 1
    then
    have b = 1^A
      using F-of-one eq-F assms(2) by auto
    then
    show ?thesis
      using 1 assms(2,4,5,7,8) ord-antisymm by blast
  next
    case 2
    then
    have b ≠ 1^A ∧ c ≠ 1^A ∧ d ≠ 1^A
      using eq-F assms(3,8) ord-antisymm ord-top by auto
    then
    have F b ⊆ F d ∧ F d ⊆ F c
      using LEMMA-3-3-3 assms(2−4,7,8) by simp
    then
    have F a = F d
      using eq-F by blast
    then
    have a ∼F d
      using assms(1,4) rel-F-equiv by simp
    then
    show ?thesis
      using assms(4) rel-F-canonical-map-def by simp
  qed
qed

lemma related-iff-same-class:
  assumes a ∈ A b ∈ A
  shows a ∼F b ⟷ π a = π b
proof
  assume a ∼F b
  then
  have a = 1^A ⟷ b = 1^A
    using assms imp-one-C imp-reflex rel-F-def by metis
  then
  have (a = 1^A ∧ b = 1^A) ∨ (a ≠ 1^A ∧ b ≠ 1^A)
    by auto
```

then
show $\pi\ a = \pi\ b$
using ‹$a \sim F\ b$› *assms rel-F-canonical-map-def rel-F-def rel-F-symm* **by** *force*
next
show $\pi\ a = \pi\ b \Longrightarrow a \sim F\ b$
using *assms*(*2*) *classes-not-empty rel-F-canonical-map-def* **by** *auto*
qed

corollary *same-F-iff-same-class*:
assumes $a \in A\ b \in A$
shows $F\ a = F\ b \longleftrightarrow \pi\ a = \pi\ b$
using *assms rel-F-equiv related-iff-same-class* **by** *auto*

end

4.4 Irreducible hoops: definition and equivalences

A totally ordered hoop is *irreducible* if it cannot be written as the ordinal sum of two nontrivial totally ordered hoops.

locale *totally-ordered-irreducible-hoop* = *totally-ordered-hoop* +
assumes *irreducible*: $\nexists\ B\ C.$
$(A = B \cup C) \wedge$
$(\{1^A\} = B \cap C) \wedge$
$(\exists\ y \in B.\ y \neq 1^A) \wedge$
$(\exists\ y \in C.\ y \neq 1^A) \wedge$
$(hoop\ B\ (*^A)\ (\rightarrow^A)\ 1^A) \wedge$
$(hoop\ C\ (*^A)\ (\rightarrow^A)\ 1^A) \wedge$
$(\forall\ x \in B - \{1^A\}.\ \forall\ y \in C.\ x *^A y = x) \wedge$
$(\forall\ x \in B - \{1^A\}.\ \forall\ y \in C.\ x \rightarrow^A y = 1^A) \wedge$
$(\forall\ x \in C.\ \forall\ y \in B.\ x \rightarrow^A y = y)$

lemma *irr-test*:
assumes *totally-ordered-hoop* $A\ PA\ RA\ a$
$\neg$*totally-ordered-irreducible-hoop* $A\ PA\ RA\ a$
shows $\exists\ B\ C.$
$(A = B \cup C) \wedge$
$(\{a\} = B \cap C) \wedge$
$(\exists\ y \in B.\ y \neq a) \wedge$
$(\exists\ y \in C.\ y \neq a) \wedge$
$(hoop\ B\ PA\ RA\ a) \wedge$
$(hoop\ C\ PA\ RA\ a) \wedge$
$(\forall\ x \in B - \{a\}.\ \forall\ y \in C.\ PA\ x\ y = x) \wedge$
$(\forall\ x \in B - \{a\}.\ \forall\ y \in C.\ RA\ x\ y = a) \wedge$
$(\forall\ x \in C.\ \forall\ y \in B.\ RA\ x\ y = y)$
using *assms* **unfolding** *totally-ordered-irreducible-hoop-def*
totally-ordered-irreducible-hoop-axioms-def
by *force*

locale *totally-ordered-one-fixed-hoop* = *totally-ordered-hoop* +

assumes *one-fixed*: $x \in A \Longrightarrow y \in A \Longrightarrow y \to^A x = x \Longrightarrow x = 1^A \vee y = 1^A$

locale *totally-ordered-wajsberg-hoop* = *totally-ordered-hoop* + *wajsberg-hoop*

context *totally-ordered-hoop*
begin

The following result can be found in [1] (see Lemma 3.5).

lemma *not-one-fixed-implies-not-irreducible*:
assumes $\neg$*totally-ordered-one-fixed-hoop* A $(*^A)$ $(\to^A)$ 1^A
shows $\neg$*totally-ordered-irreducible-hoop* A $(*^A)$ $(\to^A)$ 1^A
proof −
have $\exists\ x\ y.\ x \in A \wedge y \in A \wedge y \to^A x = x \wedge x \neq 1^A \wedge y \neq 1^A$
using *assms totally-ordered-hoop-axioms totally-ordered-one-fixed-hoop.intro totally-ordered-one-fixed-hoop-axioms.intro*
by *meson*
then
obtain b_0 c_0 **where** *witnesses*: $b_0 \in A-\{1^A\} \wedge c_0 \in A-\{1^A\} \wedge b_0 \to^A c_0 = c_0$
by *auto*
define B C **where** $B = (F\ b_0) \cup \{1^A\}$ **and** $C = A-(F\ b_0)$

have *B-mult-b0*: $b *^A b_0 = b$ **if** $b \in B-\{1^A\}$ **for** b
proof −
have *upper-bound*: $b \leq^A b_0$ **if** $b \in B-\{1^A\}$ **for** b
using *B-def F-bound witnesses that* **by** *force*
then
have $b *^A b_0 = b_0 *^A b$
using *B-def witnesses mult-comm that* **by** *simp*
also
have $\ldots = b_0 *^A (b_0 \to^A b)$
using *B-def witnesses that* **by** *fastforce*
also
have $\ldots = b *^A (b \to^A b_0)$
using *B-def witnesses that divisibility* **by** *auto*
also
have $\ldots = b$
using *B-def hoop-order-def that upper-bound witnesses* **by** *auto*
finally
show $b *^A b_0 = b$
by *auto*
qed

have *C-upper-set*: $a \in C$ **if** $a \in A$ $c \in C$ $c \leq^A a$ **for** a c
proof −
consider (1) $a \neq 1^A$
| (2) $a = 1^A$
by *auto*
then
show $a \in C$

proof(*cases*)
case *1*
then
have $a \notin C \Longrightarrow a \in F\ b_0$
using *C-def that*(*1*) **by** *blast*
then
have $a \notin C \Longrightarrow c \in F\ b_0$
using *C-def DiffD1 witnesses LEMMA-3-3-1 that* **by** *metis*
then
show *?thesis*
using *C-def that*(*2*) **by** *blast*
next
case *2*
then
show *?thesis*
using *C-def witnesses* **by** *auto*
qed
qed

have *B-union-C*: $A = B \cup C$
using *B-def C-def witnesses one-closed* **by** *auto*

moreover

have *B-inter-C*: $\{1^A\} = B \cap C$
using *B-def C-def witnesses* **by** *force*

moreover

have *B-not-trivial*: $\exists\ y \in B.\ y \neq 1^A$
proof −
have $c_0 \in B \wedge c_0 \neq 1^A$
using *B-def witnesses* **by** *auto*
then
show *?thesis*
by *auto*
qed

moreover

have *C-not-trivial*: $\exists\ y \in C.\ y \neq 1^A$
proof −
have $b_0 \in C \wedge b_0 \neq 1^A$
using *C-def witnesses* **by** *auto*
then
show *?thesis*
by *auto*
qed

moreover

have *B-mult-closed*: $a *^A b \in B$ **if** $a \in B$ $b \in B$ **for** a b
proof −
  from *that*
  consider (*1*) $a \in F\ b_0$
    | (*2*) $a = 1^A$
    using *B-def* **by** *blast*
  then
  show $a *^A b \in B$
  proof(*cases*)
    case *1*
    then
    have $a \in A \wedge a *^A b \in A \wedge (a *^A b) \leq^A a$
      using *B-union-C that mult-A mult-closed* **by** *blast*
    then
    have $a *^A b \in F\ b_0$
      using *1 witnesses LEMMA-3-3-1* **by** *metis*
    then
    show *?thesis*
      using *B-def* **by** *simp*
  next
    case *2*
    then
    show *?thesis*
      using *B-union-C that*(*2*) **by** *simp*
  qed
qed

moreover

have *B-imp-closed*: $a \to^A b \in B$ **if** $a \in B$ $b \in B$ **for** a b
proof −
  from *that*
  consider (*1*) $a = 1^A \vee b = 1^A \vee (a \in F\ b_0 \wedge b \in F\ b_0 \wedge a \to^A b = 1^A)$
    | (*2*) $a \in F\ b_0$ $b \in F\ b_0$ $a \to^A b \neq 1^A$
    using *B-def* **by** *fastforce*
  then
  show $a \to^A b \in B$
  proof(*cases*)
    case *1*
    then
    have $a \to^A b = b \vee a \to^A b = 1^A$
      using *B-union-C that imp-one-C imp-one-top* **by** *blast*
    then
    show *?thesis*
      using *B-inter-C that*(*2*) **by** *fastforce*
  next
    case *2*

then
have $a *^A b_0 = a$
using *B-def B-mult-b0 witnesses* **by** *auto*
then
have $b_0 \to^A (a \to^A b) = (a \to^A b)$
using *B-union-C witnesses that mult-comm residuation* **by** *simp*
then
have $a \to^A b \in F\ b_0$
using *2(3) B-union-C F-equiv witnesses that imp-closed* **by** *auto*
then
show *?thesis*
using *B-def* **by** *auto*
qed
qed

moreover

have *B-hoop*: *hoop* $B\ (*^A)\ (\to^A)\ 1^A$
proof
show $x *^A y \in B$ **if** $x \in B\ y \in B$ **for** $x\ y$
using *B-mult-closed that* **by** *simp*
next
show $x \to^A y \in B$ **if** $x \in B\ y \in B$ **for** $x\ y$
using *B-imp-closed that* **by** *simp*
next
show $1^A \in B$
using *B-def* **by** *simp*
next
show $x *^A y = y *^A x$ **if** $x \in B\ y \in B$ **for** $x\ y$
using *B-union-C mult-comm that* **by** *simp*
next
show $x *^A (y *^A z) = (x *^A y) *^A z$ **if** $x \in B\ y \in B\ z \in B$ **for** $x\ y\ z$
using *B-union-C mult-assoc that* **by** *simp*
next
show $x *^A 1^A = x$ **if** $x \in B$ **for** x
using *B-union-C that* **by** *simp*
next
show $x \to^A x = 1^A$ **if** $x \in B$ **for** x
using *B-union-C that* **by** *simp*
next
show $x *^A (x \to^A y) = y *^A (y \to^A x)$ **if** $x \in B\ y \in B$ **for** $x\ y$
using *B-union-C divisibility that* **by** *simp*
next
show $x \to^A (y \to^A z) = (x *^A y) \to^A z$ **if** $x \in B\ y \in B\ z \in B$ **for** $x\ y\ z$
using *B-union-C residuation that* **by** *simp*
qed

moreover

```
have C-imp-B: c →^A b = b if b ∈ B c ∈ C for b c
proof −
  from that
  consider (1) b ∈ F b_0 c ≠ 1^A
    | (2) b = 1^A ∨ c = 1^A
    using B-def by blast
  then
  show c →^A b = b
  proof(cases)
    case 1
    have b_0 →^A ((c →^A b) →^A b) = (c →^A b) →^A (b_0 →^A b)
      using B-union-C witnesses that imp-closed swap by simp
    also
    have ... = (c →^A b) →^A b
      using 1(1) witnesses by auto
    finally
    have (c →^A b) →^A b ∈ F b_0 if (c →^A b) →^A b ≠ 1^A
     using B-union-C F-equiv witnesses ‹b ∈ B› ‹c ∈ C› that imp-closed by auto
    moreover
    have c ≤^A ((c →^A b) →^A b)
      using B-union-C that ord-C by simp
    ultimately
    have (c →^A b) →^A b = 1^A
      using B-def B-union-C C-def C-upper-set that(2) by blast
    moreover
    have b →^A (c →^A b) = 1^A
      using B-union-C that imp-A by simp
    ultimately
    show ?thesis
      using B-union-C that imp-closed ord-antisymm-equiv by blast
  next
    case 2
    then
    show ?thesis
      using B-union-C that imp-one-C imp-one-top by auto
  qed
qed

moreover

have B-imp-C: b →^A c = 1^A if b ∈ B−{1^A} c ∈ C for b c
proof −
  from that
  have b ≤^A c ∨ c ≤^A b
    using total-order B-union-C by blast
  moreover
  have c →^A b = b
    using C-imp-B that by simp
  ultimately
```

show $b \to^A c = 1^A$
  using *that*(*1*) *hoop-order-def* **by** *force*
qed

moreover

have *B-mult-C*: $b *^A c = b$ **if** $b \in B-\{1^A\}$ $c \in C$ **for** b c
proof −
  have $b = b *^A 1^A$
    using *that*(*1*) *B-union-C* **by** *fastforce*
  also
  have $\ldots = b *^A (b \to^A c)$
    using *B-imp-C that* **by** *blast*
  also
  have $\ldots = c *^A (c \to^A b)$
    using *that divisibility B-union-C* **by** *simp*
  also
  have $\ldots = c *^A b$
    using *C-imp-B that* **by** *auto*
  finally
  show $b *^A c = b$
    using *that mult-comm B-union-C* **by** *auto*
qed

moreover

have *C-mult-closed*: $c *^A d \in C$ **if** $c \in C$ $d \in C$ **for** c d
proof −
  consider (*1*) $c \neq 1^A$ $d \neq 1^A$
    | (*2*) $c = 1^A \vee d = 1^A$
    by *auto*
  then
  show $c *^A d \in C$
  proof(*cases*)
    case *1*
    have $c *^A d \in F\ b_0$ **if** $c *^A d \notin C$
      using *C-def* ‹$c \in C$› ‹$d \in C$› *mult-closed that* **by** *force*
    then
    have $c \to^A (c *^A d) \in F\ b_0$ **if** $c *^A d \notin C$
      using *B-def C-imp-B* ‹$c \in C$› *that* **by** *simp*
    moreover
    have $d \leq^A (c \to^A (c *^A d))$
      using *C-def DiffD1 that ord-reflex ord-residuation residuation*
          mult-closed mult-comm
      by *metis*
    moreover
    have $c \to^A (c *^A d) \in A \wedge d \in A$
      using *C-def Diff-iff that imp-closed mult-closed* **by** *metis*
    ultimately

have $d \in F\ b_0$ **if** $c *^A d \notin C$
using *witnesses LEMMA-3-3-1 that* **by** *blast*
then
show *?thesis*
using *C-def that(2)* **by** *blast*
next
case *2*
then
show *?thesis*
using *B-union-C that mult-neutr mult-neutr-2* **by** *auto*
qed
qed

moreover

have *C-imp-closed*: $c \rightarrow^A d \in C$ **if** $c \in C$ $d \in C$ **for** $c\ d$
using *C-upper-set imp-closed ord-A B-union-C that* **by** *blast*

moreover

have *C-hoop*: *hoop* $C\ (*^A)\ (\rightarrow^A)\ 1^A$
proof
show $x *^A y \in C$ **if** $x \in C$ $y \in C$ **for** $x\ y$
using *C-mult-closed that* **by** *simp*
next
show $x \rightarrow^A y \in C$ **if** $x \in C$ $y \in C$ **for** $x\ y$
using *C-imp-closed that* **by** *simp*
next
show $1^A \in C$
using *B-inter-C* **by** *auto*
next
show $x *^A y = y *^A x$ **if** $x \in C$ $y \in C$ **for** $x\ y$
using *B-union-C mult-comm that* **by** *simp*
next
show $x *^A (y *^A z) = (x *^A y) *^A z$ **if** $x \in C$ $y \in C$ $z \in C$ **for** $x\ y\ z$
using *B-union-C mult-assoc that* **by** *simp*
next
show $x *^A 1^A = x$ **if** $x \in C$ **for** x
using *B-union-C that* **by** *simp*
next
show $x \rightarrow^A x = 1^A$ **if** $x \in C$ **for** x
using *B-union-C that* **by** *simp*
next
show $x *^A (x \rightarrow^A y) = y *^A (y \rightarrow^A x)$ **if** $x \in C$ $y \in C$ **for** $x\ y$
using *B-union-C divisibility that* **by** *simp*
next
show $x \rightarrow^A (y \rightarrow^A z) = (x *^A y) \rightarrow^A z$ **if** $x \in C$ $y \in C$ $z \in C$ **for** $x\ y\ z$
using *B-union-C residuation that* **by** *simp*
qed

ultimately

have $\exists\ B\ C.$
$(A = B \cup C) \wedge$
$(\{1^A\} = B \cap C) \wedge$
$(\exists\ y \in B.\ y \neq 1^A) \wedge$
$(\exists\ y \in C.\ y \neq 1^A) \wedge$
$(hoop\ B\ (*^A)\ (\rightarrow^A)\ 1^A) \wedge$
$(hoop\ C\ (*^A)\ (\rightarrow^A)\ 1^A) \wedge$
$(\forall\ x \in B-\{1^A\}.\ \forall\ y \in C.\ x *^A y = x) \wedge$
$(\forall\ x \in B-\{1^A\}.\ \forall\ y \in C.\ x \rightarrow^A y = 1^A) \wedge$
$(\forall\ x \in C.\ \forall\ y \in B.\ x \rightarrow^A y = y)$
by (*smt* (*verit*))
then
show *?thesis*
using *totally-ordered-irreducible-hoop.irreducible* **by** (*smt* (*verit*))
qed

Next result can be found in [2] (see Proposition 2.2).

lemma *one-fixed-implies-wajsberg*:
assumes *totally-ordered-one-fixed-hoop* $A\ (*^A)\ (\rightarrow^A)\ 1^A$
shows *totally-ordered-wajsberg-hoop* $A\ (*^A)\ (\rightarrow^A)\ 1^A$
proof
have $(a \rightarrow^A b) \rightarrow^A b = (b \rightarrow^A a) \rightarrow^A a$ **if** $a \in A\ b \in A\ a <^A b$ **for** $a\ b$
proof −
from *that*
have $(((b \rightarrow^A a) \rightarrow^A a) \rightarrow^A b) \rightarrow^A (b \rightarrow^A a) = b \rightarrow^A a \wedge b \rightarrow^A a \neq 1^A$
using *imp-D ord-D* **by** *simp*
then
have $((b \rightarrow^A a) \rightarrow^A a) \rightarrow^A b = 1^A$
using *assms that*(*1*,*2*) *imp-closed totally-ordered-one-fixed-hoop.one-fixed*
by *metis*
moreover
have $b \rightarrow^A ((b \rightarrow^A a) \rightarrow^A a) = 1^A$
using *hoop-order-def that*(*1*,*2*) *ord-C* **by** *simp*
ultimately
have $(b \rightarrow^A a) \rightarrow^A a = b$
using *imp-closed ord-antisymm-equiv hoop-axioms that*(*1*,*2*) **by** *metis*
also
have $\ldots = (a \rightarrow^A b) \rightarrow^A b$
using *hoop-order-def hoop-order-strict-def that*(*2*,*3*) *imp-one-C* **by** *force*
finally
show $(a \rightarrow^A b) \rightarrow^A b = (b \rightarrow^A a) \rightarrow^A a$
by *auto*
qed
then
show $(x \rightarrow^A y) \rightarrow^A y = (y \rightarrow^A x) \rightarrow^A x$ **if** $x \in A\ y \in A$ **for** $x\ y$
using *total-order hoop-order-strict-def that* **by** *metis*

qed

The proof of the following result can be found in [1] (see Theorem 3.6).

lemma *not-irreducible-implies-not-wajsberg*:
assumes $\neg$*totally-ordered-irreducible-hoop* A $(*^A)$ $(\to^A)$ 1^A
shows $\neg$*totally-ordered-wajsberg-hoop* A $(*^A)$ $(\to^A)$ 1^A
proof −
have $\exists\ B\ C.$
$(A = B \cup C) \wedge$
$(\{1^A\} = B \cap C) \wedge$
$(\exists\ y \in B.\ y \neq 1^A) \wedge$
$(\exists\ y \in C.\ y \neq 1^A) \wedge$
$(hoop\ B\ (*^A)\ (\to^A)\ 1^A) \wedge$
$(hoop\ C\ (*^A)\ (\to^A)\ 1^A) \wedge$
$(\forall\ x \in B-\{1^A\}.\ \forall\ y \in C.\ x *^A y = x) \wedge$
$(\forall\ x \in B-\{1^A\}.\ \forall\ y \in C.\ x \to^A y = 1^A) \wedge$
$(\forall\ x \in C.\ \forall\ y \in B.\ x \to^A y = y)$
using *irr-test*[*OF totally-ordered-hoop-axioms*] *assms* **by** *auto*
then
obtain B C **where** H:
$(A = B \cup C) \wedge$
$(\{1^A\} = B \cap C) \wedge$
$(\exists\ y \in B.\ y \neq 1^A) \wedge$
$(\exists\ y \in C.\ y \neq 1^A) \wedge$
$(\forall\ x \in B-\{1^A\}.\ \forall\ y \in C.\ x \to^A y = 1^A) \wedge$
$(\forall\ x \in C.\ \forall\ y \in B.\ x \to^A y = y)$
by *blast*
then
obtain b c **where** *assms*: $b \in B-\{1^A\} \wedge c \in C-\{1^A\}$
by *auto*
then
have $b \to^A c = 1^A$
using H **by** *simp*
then
have $(b \to^A c) \to^A c = c$
using H *assms imp-one-C* **by** *blast*
moreover
have $(c \to^A b) \to^A b = 1^A$
using *assms* H **by** *force*
ultimately
have $(b \to^A c) \to^A c \neq (c \to^A b) \to^A b$
using *assms* **by** *force*
moreover
have $b \in A \wedge c \in A$
using *assms* H **by** *blast*
ultimately
show *?thesis*
using *totally-ordered-wajsberg-hoop.axioms(2) wajsberg-hoop.T* **by** *meson*
qed

Summary of all results in this subsection:

theorem *one-fixed-equivalent-to-wajsberg*:
  shows *totally-ordered-one-fixed-hoop A* $(*^A)$ $(\to^A)$ $1^A \equiv$
  totally-ordered-wajsberg-hoop A $(*^A)$ $(\to^A)$ 1^A
  using *not-irreducible-implies-not-wajsberg not-one-fixed-implies-not-irreducible one-fixed-implies-wajsberg*
  by *linarith*

theorem *wajsberg-equivalent-to-irreducible*:
  shows *totally-ordered-wajsberg-hoop A* $(*^A)$ $(\to^A)$ $1^A \equiv$
  totally-ordered-irreducible-hoop A $(*^A)$ $(\to^A)$ 1^A
  using *not-irreducible-implies-not-wajsberg not-one-fixed-implies-not-irreducible one-fixed-implies-wajsberg*
  by *linarith*

theorem *irreducible-equivalent-to-one-fixed*:
  shows *totally-ordered-irreducible-hoop A* $(*^A)$ $(\to^A)$ $1^A \equiv$
  totally-ordered-one-fixed-hoop A $(*^A)$ $(\to^A)$ 1^A
  using *one-fixed-equivalent-to-wajsberg wajsberg-equivalent-to-irreducible*
  by *simp*

end

4.5 Decomposition

locale *tower-of-irr-hoops* = *tower-of-hoops* +
  assumes *family-of-irr-hoops*: $i \in I \Longrightarrow$
  totally-ordered-irreducible-hoop $(\mathbb{A}_i)$ $(*^i)$ $(\to^i)$ 1^S

locale *tower-of-nontrivial-irr-hoops* = *tower-of-irr-hoops* +
  assumes *nontrivial*: $i \in I \Longrightarrow \exists\ x \in \mathbb{A}_i.\ x \neq 1^S$

context *totally-ordered-hoop*
begin

4.5.1 Definition of index set I

definition *index-set* :: $('a\ set)\ set$ (I)
  where $I = \{y.\ (\exists\ x \in A.\ \pi\ x = y)\}$

lemma *indexes-subsets*:
  assumes $i \in I$
  shows $i \subseteq A$
  using *index-set-def assms rel-F-canonical-map-def* **by** *auto*

lemma *indexes-not-empty*:
  assumes $i \in I$
  shows $i \neq \emptyset$
  using *index-set-def assms classes-not-empty* **by** *blast*

lemma *indexes-disjoint*:
  assumes $i \in I$ $j \in I$ $i \neq j$
  shows $i \cap j = \emptyset$
proof –
  obtain a b **where** $a \in A \wedge b \in A \wedge a \neq b \wedge i = \pi\ a \wedge j = \pi\ b$
    using *index-set-def assms* **by** *auto*
  then
  show *?thesis*
    using *assms*(*3*) *classes-disjoint* **by** *auto*
qed

lemma *indexes-cover*: $A = \{x.\ \exists\ i \in I.\ x \in i\}$
  using *classes-subsets classes-not-empty index-set-def* **by** *auto*

lemma *indexes-class-of-elements*:
  assumes $i \in I$ $a \in A$ $a \in i$
  shows $\pi\ a = i$
proof –
  obtain c **where** *class-element*: $c \in A \wedge i = \pi\ c$
    using *assms*(*1*) *index-set-def* **by** *auto*
  then
  have $a \sim F\ c$
    using *assms*(*3*) *rel-F-canonical-map-def rel-F-symm* **by** *auto*
  then
  show *?thesis*
    using *assms*(*2*) *class-element related-iff-same-class* **by** *simp*
qed

lemma *indexes-convex*:
  assumes $i \in I$ $a \in i$ $b \in i$ $d \in A$ $a \leq^A d$ $d \leq^A b$
  shows $d \in i$
proof –
  have $a \in A \wedge b \in A \wedge d \in A \wedge i = \pi\ a$
    using *assms*(*1*–*4*) *indexes-class-of-elements indexes-subsets* **by** *blast*
  then
  show *?thesis*
    using *assms*(*2*–*6*) *classes-convex* **by** *auto*
qed

4.5.2 Definition of total partial order over I

Since each equivalence class is convex, $(\leq^A)$ induces a total order on I.

function *index-order* :: ('*a set*) ⇒ ('*a set*) ⇒ *bool* (**infix** $\leq^I$ *60*) **where**
  $x \leq^I y = ((x = y) \vee (\forall\ v \in x.\ \forall\ w \in y.\ v \leq^A w))$ **if** $x \in I$ $y \in I$
| $x \leq^I y =$ *undefined* **if** $x \notin I \vee y \notin I$
  by *auto*
termination by *lexicographic-order*

definition *index-order-strict* (**infix** $<^I$ *60*)
  where $x <^I y = (x \leq^I y \wedge x \neq y)$

lemma *index-ord-reflex*:
  assumes $i \in I$
  shows $i \leq^I i$
  using *assms* **by** *simp*

lemma *index-ord-antisymm*:
  assumes $i \in I$ $j \in I$ $i \leq^I j$ $j \leq^I i$
  shows $i = j$
proof −
  have $i = j \vee (\forall\ a \in i.\ \forall\ b \in j.\ a \leq^A b \wedge b \leq^A a)$
    using *assms* **by** *auto*
  then
  have $i = j \vee (\forall\ a \in i.\ \forall\ b \in j.\ a = b)$
    using *assms*(*1*,*2*) *indexes-subsets insert-Diff insert-subset ord-antisymm*
    by *metis*
  then
  show *?thesis*
    using *assms*(*1*,*2*) *indexes-not-empty* **by** *force*
qed

lemma *index-ord-trans*:
  assumes $i \in I$ $j \in I$ $k \in I$ $i \leq^I j$ $j \leq^I k$
  shows $i \leq^I k$
proof −
  consider (*1*) $i \neq j$ $j \neq k$
    | (*2*) $i = j \vee j = k$
    by *auto*
  then
  show $i \leq^I k$
  proof(*cases*)
    case *1*
    then
    have $(\forall\ a \in i.\ \forall\ b \in j.\ a \leq^A b) \wedge (\forall\ b \in j.\ \forall\ c \in k.\ b \leq^A c)$
      using *assms* **by** *force*
    moreover
    have $j \neq \emptyset$
      using *assms*(*2*) *indexes-not-empty* **by** *simp*
    ultimately
    have $\forall\ a \in i.\ \forall\ c \in k.\ \exists\ b \in j.\ a \leq^A b \wedge b \leq^A c$
      using *all-not-in-conv* **by** *meson*
    then
    have $\forall\ a \in i.\ \forall\ c \in k.\ a \leq^A c$
      using *assms indexes-subsets ord-trans subsetD* **by** *metis*
    then
    show *?thesis*
      using *assms*(*1*,*3*) **by** *simp*

```
  next
    case 2
    then
    show ?thesis
      using assms(4,5) by auto
  qed
qed

lemma index-order-total :
  assumes i ∈ I j ∈ I ¬(j ≤^I i)
  shows i ≤^I j
proof −
  have i ≠ j
    using assms(1,3) by auto
  then
  have disjoint: i ∩ j = ∅
    using assms(1,2) indexes-disjoint by simp
  moreover
  have ∃ x ∈ j. ∃ y ∈ i. ¬(x ≤^A y)
    using assms index-order.simps(1) by blast
  moreover
  have subsets: i ⊆ A ∧ j ⊆ A
    using assms indexes-subsets by simp
  ultimately
  have ∃ x ∈ j. ∃ y ∈ i. y <^A x
    using total-order hoop-order-strict-def insert-absorb insert-subset by metis
  then
  obtain a_i a_j where witnesses: a_i ∈ i ∧ a_j ∈ j ∧ a_i <^A a_j
    using assms(1,2) total-order hoop-order-strict-def indexes-subsets by metis
  then
  have a ≤^A b if a ∈ i b ∈ j for a b
  proof
    from that
    consider (1) a_i ≤^A a a_j ≤^A b
      | (2) a <^A a_i b <^A a_j
      | (3) a_i ≤^A a b <^A a_j
      | (4) a <^A a_i a_j ≤^A b
      using total-order hoop-order-strict-def subset-eq subsets witnesses by metis
    then
    show a ≤^A b
    proof(cases)
      case 1
      then
      have a_i ≤^A a_j ∧ a_j ≤^A b ∧ b ≤^A a if b <^A a
        using hoop-order-strict-def that witnesses by blast
      then
      have a_i ≤^A b ∧ b ≤^A a if b <^A a
        using ‹b ∈ j› in-mono ord-trans subsets that witnesses by meson
      then
```

have $b \in i$ **if** $b <^A a$
  using *assms(1)* ‹$a \in i$› ‹$b \in j$› *indexes-convex subsets that witnesses*
  by *blast*
then
show $a \leq^A b$
  using *disjoint disjoint-iff-not-equal hoop-order-strict-def subset-eq subsets that total-order*
  by *metis*
next
case *2*
then
have $b \leq^A a \wedge a \leq^A a_i \wedge a_i \leq^A a_j$ **if** $b <^A a$
  using *hoop-order-strict-def that witnesses* **by** *blast*
then
have $b \leq^A a \wedge a \leq^A a_j$ **if** $b <^A a$
  using ‹$a \in i$› *ord-trans subset-eq subsets that witnesses* **by** *metis*
then
have $a \in j$ **if** $b <^A a$
  using *assms(2)* ‹$a \in i$› ‹$b \in j$› *indexes-convex subsets that witnesses*
  by *blast*
then
show $a \leq^A b$
  using *disjoint disjoint-iff-not-equal hoop-order-strict-def subset-eq subsets that total-order*
  by *metis*
next
case *3*
have $b \leq^A a_i \wedge a_i \leq^A a_j$ **if** $b \leq^A a_i$
  using *hoop-order-strict-def that witnesses* **by** *auto*
then
have $a_i \in j$ **if** $b \leq^A a_i$
  using *assms(2)* ‹$b \in j$› *indexes-convex subsets that witnesses* **by** *blast*
moreover
have $a_i \notin j$
  using *disjoint witnesses* **by** *blast*
ultimately
have $a_i <^A b$
  using *total-order hoop-order-strict-def* ‹$b \in j$› *subsets witnesses* **by** *blast*
then
have $a_i \leq^A b \wedge b \leq^A a$ **if** $b <^A a$
  using *hoop-order-strict-def that* **by** *auto*
then
have $b \in i$ **if** $b <^A a$
  using *assms(1)* ‹$a \in i$› ‹$b \in j$› *indexes-convex subsets that witnesses*
  by *blast*
then
show $a \leq^A b$
  using *disjoint disjoint-iff-not-equal hoop-order-strict-def subset-eq subsets that total-order*

by *metis*
next
case *4*
then
show $a \leq^A b$
using *hoop-order-strict-def in-mono ord-trans subsets that witnesses*
by *meson*
qed
qed
then
show $i \leq^I j$
using *assms* **by** *simp*
qed

sublocale *total-poset-on* I $(\leq^I)$ $(<^I)$
proof
show $I \neq \emptyset$
using *indexes-cover* **by** *auto*
next
show *reflp-on* I $(\leq^I)$
using *index-ord-reflex reflp-onI* **by** *blast*
next
show *antisymp-on* I $(\leq^I)$
using *antisymp-on-def index-ord-antisymm* **by** *blast*
next
show *transp-on* I $(\leq^I)$
using *index-ord-trans transp-on-def* **by** *blast*
next
show $x <^I y = (x \leq^I y \wedge x \neq y)$ **if** $x \in I$ $y \in I$ **for** x y
using *index-order-strict-def* **by** *auto*
next
show *totalp-on* I $(\leq^I)$
using *index-order-total totalp-onI* **by** *metis*
qed

4.5.3 Definition of universes

definition *universes* :: $'a\ set \Rightarrow 'a\ set$ (UNI_A)
where $UNI_A\ x = x \cup \{1^A\}$

abbreviation (*uniA-i*)
uniA-i :: $['a\ set] \Rightarrow ('a\ set)$ $((\mathbb{A}(\_))\ [61]\ 60)$
where $\mathbb{A}_i \equiv UNI_A\ i$

abbreviation (*uniA-pi*)
uniA-pi :: $['a] \Rightarrow ('a\ set)$ $((\mathbb{A}_\pi\ (\_))\ [61]\ 60)$
where $\mathbb{A}_{\pi x} \equiv UNI_A\ (\pi\ x)$

abbreviation (*uniA-pi-one*)

uniA-pi-one :: *'a set* $((\mathbb{A}_{\pi 1 A})$ *60*)
where $\mathbb{A}_{\pi 1 A} \equiv UNI_A\ (\pi\ 1^A)$

lemma *universes-subsets*:
assumes $i \in I\ a \in \mathbb{A}_i$
shows $a \in A$
using *assms universes-def indexes-subsets one-closed* **by** *fastforce*

lemma *universes-not-empty*:
assumes $i \in I$
shows $\mathbb{A}_i \neq \emptyset$
using *universes-def* **by** *simp*

lemma *universes-almost-disjoint*:
assumes $i \in I\ j \in I\ i \neq j$
shows $(\mathbb{A}_i) \cap (\mathbb{A}_j) = \{1^A\}$
using *assms indexes-disjoint universes-def* **by** *auto*

lemma *universes-cover*: $A = \{x.\ \exists\ i \in I.\ x \in \mathbb{A}_i\}$
using *one-closed indexes-cover universes-def* **by** *auto*

lemma *universes-aux*:
assumes $i \in I\ a \in i$
shows $\mathbb{A}_i = \pi\ a \cup \{1^A\}$
using *assms universes-def universes-subsets indexes-class-of-elements* **by** *force*

4.5.4 Universes are subhoops of A

lemma *universes-one-closed*:
assumes $i \in I$
shows $1^A \in \mathbb{A}_i$
using *universes-def* **by** *auto*

lemma *universes-mult-closed*:
assumes $i \in I\ a \in \mathbb{A}_i\ b \in \mathbb{A}_i$
shows $a *^A\ b \in \mathbb{A}_i$
proof −
consider (*1*) $a \neq 1^A\ b \neq 1^A$
| (*2*) $a = 1^A \vee b = 1^A$
by *auto*
then
show *?thesis*
proof(*cases*)
case *1*
then
have *UNI-def*: $\mathbb{A}_i = \pi\ a \cup \{1^A\} \wedge \mathbb{A}_i = \pi\ b \cup \{1^A\}$
using *assms universes-def universes-subsets indexes-class-of-elements*
by *simp*
then

```
    have π a = π b
      using 1 assms universes-def universes-subsets indexes-class-of-elements
      by force
    then
    have F a = F b
      using assms universes-subsets rel-F-equiv related-iff-same-class by meson
    then
    have F (a *^A b) = F a
      using 1 LEMMA-3-3-2 assms universes-subsets by blast
    then
    have π a = π (a *^A b)
      using assms universes-subsets mult-closed rel-F-equiv related-iff-same-class
      by metis
    then
    show ?thesis
      using UNI-def UnI1 assms classes-not-empty universes-subsets mult-closed
      by metis
  next
    case 2
    then
    show ?thesis
      using assms universes-subsets by auto
  qed
qed

lemma universes-imp-closed:
  assumes i ∈ I a ∈ 𝔸_i b ∈ 𝔸_i
  shows a →^A b ∈ 𝔸_i
proof −
  from assms
  consider (1) a ≠ 1^A b ≠ 1^A b <^A a
    | (2) a = 1^A ∨ b = 1^A ∨ (a ≠ 1^A ∧ b ≠ 1^A ∧ a ≤^A b)
    using total-order universes-subsets hoop-order-strict-def by auto
  then
  show ?thesis
  proof(cases)
    case 1
    then
    have UNI-def: 𝔸_i = π a ∪ {1^A} ∧ 𝔸_i = π b ∪ {1^A}
      using assms universes-def universes-subsets indexes-class-of-elements
      by simp
    then
    have π a = π b
      using 1 assms universes-def universes-subsets indexes-class-of-elements
      by force
    then
    have F a = F b
      using assms universes-subsets rel-F-equiv related-iff-same-class by simp
    then
```

have $F\ (a \to^A\ b) = F\ a$
using *1 LEMMA-3-3-6 assms universes-subsets* **by** *simp*
then
have $\pi\ a = \pi\ (a \to^A\ b)$
using *assms universes-subsets imp-closed same-F-iff-same-class* **by** *simp*
then
show *?thesis*
using *UNI-def UnI1 assms classes-not-empty universes-subsets imp-closed*
by *metis*
next
case *2*
then
show *?thesis*
using *assms universes-subsets universes-one-closed hoop-order-def imp-one-A imp-one-C*
by *auto*
qed
qed

4.5.5 Universes are irreducible hoops

lemma *universes-one-fixed*:
assumes $i \in I\ a \in \mathbb{A}_i\ b \in \mathbb{A}_i\ a \to^A\ b = b$
shows $a = 1^A \vee b = 1^A$
proof −
from *assms*
have $\pi\ a = \pi\ b$ **if** $a \neq 1^A\ b \neq 1^A$
using *universes-def universes-subsets indexes-class-of-elements that* **by** *force*
then
have $F\ a = F\ b$ **if** $a \neq 1^A\ b \neq 1^A$
using *assms(1−3) universes-subsets same-F-iff-same-class that* **by** *blast*
then
have $b = 1^A$ **if** $a \neq 1^A\ b \neq 1^A$
using *F-equiv assms universes-subsets fixed-points.cases imp-reflex that* **by** *metis*
then
show *?thesis*
by *blast*
qed

corollary *universes-one-fixed-hoops*:
assumes $i \in I$
shows *totally-ordered-one-fixed-hoop* $(\mathbb{A}_i)\ (*^A)\ (\to^A)\ 1^A$
proof
show $x *^A\ y \in \mathbb{A}_i$ **if** $x \in \mathbb{A}_i\ y \in \mathbb{A}_i$ **for** $x\ y$
using *assms universes-mult-closed that* **by** *simp*
next
show $x \to^A\ y \in \mathbb{A}_i$ **if** $x \in \mathbb{A}_i\ y \in \mathbb{A}_i$ **for** $x\ y$
using *assms universes-imp-closed that* **by** *simp*
next

show $1^A \in \mathbb{A}_i$
using *assms universes-one-closed* **by** *simp*
next
show $x *^A y = y *^A x$ **if** $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $x\ y$
using *assms universes-subsets mult-comm that* **by** *simp*
next
show $x *^A (y *^A z) = (x *^A y) *^A z$ **if** $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $x\ y\ z$
using *assms universes-subsets mult-assoc that* **by** *simp*
next
show $x *^A 1^A = x$ **if** $x \in \mathbb{A}_i$ **for** x
using *assms universes-subsets that* **by** *simp*
next
show $x \to^A x = 1^A$ **if** $x \in \mathbb{A}_i$ **for** x
using *assms universes-subsets that* **by** *simp*
next
show $x *^A (x \to^A y) = y *^A (y \to^A x)$ **if** $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $x\ y$
using *assms divisibility universes-subsets that* **by** *simp*
next
show $x \to^A (y \to^A z) = (x *^A y) \to^A z$ **if** $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $x\ y\ z$
using *assms universes-subsets residuation that* **by** *simp*
next
show $x \leq^A y \vee y \leq^A x$ **if** $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $x\ y$
using *assms total-order universes-subsets that* **by** *simp*
next
show $x = 1^A \vee y = 1^A$ **if** $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $y \to^A x = x$ **for** $x\ y$
using *assms universes-one-fixed that* **by** *blast*
qed

corollary *universes-irreducible-hoops*:
assumes $i \in I$
shows *totally-ordered-irreducible-hoop* $(\mathbb{A}_i)$ $(*^A)$ $(\to^A)$ 1^A
using *assms universes-one-fixed-hoops totally-ordered-hoop.irreducible-equivalent-to-one-fixed totally-ordered-one-fixed-hoop.axioms(1)*
by *metis*

4.5.6 Some useful results

lemma *index-aux*:
assumes $i \in I$ $j \in I$ $i <^I j$ $a \in (\mathbb{A}_i) - \{1^A\}$ $b \in (\mathbb{A}_j) - \{1^A\}$
shows $a <^A b \wedge \neg(a \sim F\ b)$
proof −
have *noteq*: $i \neq j \wedge x \leq^A y$ **if** $x \in i$ $y \in j$ **for** $x\ y$
using *assms that index-order-strict-def* **by** *fastforce*
moreover
have *ij-def*: $i = \pi\ a \wedge j = \pi\ b$
using *UnE assms universes-def universes-subsets indexes-class-of-elements*
by *auto*
ultimately
have $a <^A b$

using *assms*(*1*,*2*,*4*,*5*) *classes-not-empty universes-subsets hoop-order-strict-def*
by *blast*
moreover
have $i = j$ **if** $a \sim F\ b$
using *assms*(*1*,*2*,*4*,*5*) *that universes-subsets ij-def related-iff-same-class* **by** *auto*
ultimately
show *?thesis*
using *assms*(*2*,*3*) *trichotomy* **by** *blast*
qed

lemma *different-indexes-mult*:
assumes $i \in I\ j \in I\ i <^I j\ a \in (\mathbb{A}_i)-\{1^A\}\ b \in (\mathbb{A}_j)-\{1^A\}$
shows $a *^A b = a$
proof −
have $a <^A b \wedge \neg(a \sim F\ b)$
using *assms index-aux* **by** *blast*
then
have $a <^A b \wedge F\ a \neq F\ b$
using *DiffD1 assms*(*1*,*2*,*4*,*5*) *universes-subsets rel-F-equiv* **by** *meson*
then
have $a <^A b \wedge a *^A b = a \wedge^A b$
using *DiffD1 LEMMA-3-3-5 assms*(*1*,*2*,*4*,*5*) *universes-subsets* **by** *auto*
then
show *?thesis*
using *assms*(*1*,*2*,*4*,*5*) *universes-subsets hoop-order-strict-def inf-order* **by** *auto*
qed

lemma *different-indexes-imp-1*:
assumes $i \in I\ j \in I\ i <^I j\ a \in (\mathbb{A}_i)-\{1^A\}\ b \in (\mathbb{A}_j)-\{1^A\}$
shows $a \to^A b = 1^A$
proof −
have $x \leq^A y$ **if** $x \in i\ y \in j$ **for** $x\ y$
using *assms*(*1*−*3*) *index-order-strict-def that* **by** *fastforce*
moreover
have $a \in i \wedge b \in j$
using *assms*(*4*,*5*) *assms*(*5*) *universes-def* **by** *auto*
ultimately
show *?thesis*
using *hoop-order-def* **by** *auto*
qed

lemma *different-indexes-imp-2* :
assumes $i \in I\ j \in I\ i <^I j\ a \in (\mathbb{A}_j)-\{1^A\}\ b \in (\mathbb{A}_i)-\{1^A\}$
shows $a \to^A b = b$
proof −
have $b <^A a \wedge \neg(b \sim F\ a)$
using *assms index-aux* **by** *blast*
then
have $b <^A a \wedge F\ b \neq F\ a$

using *DiffD1 assms(1,2,4,5) universes-subsets rel-F-equiv* **by** *metis*
then
have $b \in F\ a$
using *LEMMA-3-3-4 assms(1,2,4,5) universes-subsets* **by** *simp*
then
show *?thesis*
using *assms(2,4,5) universes-subsets* **by** *fastforce*
qed

4.5.7 Definition of multiplications, implications and one

definition *mult-map* :: $'a\ set \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ (MUL_A)
where $MUL_A\ x = (*^A)$

definition *imp-map* :: $'a\ set \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ (IMP_A)
where $IMP_A\ x = (\rightarrow^A)$

definition *sum-one* :: $'a$ (1^S)
where $1^S = 1^A$

abbreviation (*multA-i*)
multA-i :: $['a\ set] \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ $((*(^-))\ [50]\ 60)$
where $*^i \equiv MUL_A\ i$

abbreviation (*impA-i*)
impA-i:: $['a\ set] \Rightarrow ('a \Rightarrow 'a \Rightarrow 'a)$ $((\rightarrow(^-))\ [50]\ 60)$
where $\rightarrow^i \equiv IMP_A\ i$

abbreviation (*multA-i-xy*)
multA-i-xy :: $['a, 'a\ set, 'a] \Rightarrow 'a$ $(((\text{-})/\ *(^-)\ /\ (\text{-}))\ [61,\ 50,\ 61]\ 60)$
where $x *^i y \equiv MUL_A\ i\ x\ y$

abbreviation (*impA-i-xy*)
impA-i-xy :: $['a, 'a\ set, 'a] \Rightarrow 'a$ $(((\text{-})/\ \rightarrow(^-)\ /\ (\text{-}))\ [61,\ 50,\ 61]\ 60)$
where $x \rightarrow^i y \equiv IMP_A\ i\ x\ y$

abbreviation (*ord-i-xy*)
ord-i-xy :: $['a, 'a\ set, 'a] \Rightarrow bool$ $(((\text{-})/\ \leq(^-)\ /\ (\text{-}))\ [61,\ 50,\ 61]\ 60)$
where $x \leq^i y \equiv$ *hoop.hoop-order* $(IMP_A\ i)\ 1^S\ x\ y$

4.5.8 Main result

We prove the main result: a totally ordered hoop is equal to an ordinal sum of a tower of irreducible hoops.

sublocale *A-SUM*: *tower-of-irr-hoops* $I\ (\leq^I)\ (<^I)\ UNI_A\ MUL_A\ IMP_A\ 1^S$
proof
show $(\mathbb{A}_i) \cap (\mathbb{A}_j) = \{1^S\}$ **if** $i \in I\ j \in I\ i \neq j$ **for** $i\ j$
using *universes-almost-disjoint sum-one-def that* **by** *simp*
next

show $x *^i y \in \mathbb{A}_i$ **if** $i \in I$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *universes-mult-closed mult-map-def that* **by** *simp*
next
show $x \rightarrow^i y \in \mathbb{A}_i$ **if** $i \in I$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *universes-imp-closed imp-map-def that* **by** *simp*
next
show $1^S \in \mathbb{A}_i$ **if** $i \in I$ **for** i
using *universes-one-closed sum-one-def that* **by** *simp*
next
show $x *^i y = y *^i x$ **if** $i \in I$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *universes-subsets mult-comm mult-map-def that* **by** *simp*
next
show $x *^i (y *^i z) = (x *^i y) *^i z$
if $i \in I$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $i\ x\ y\ z$
using *universes-subsets mult-assoc mult-map-def that* **by** *simp*
next
show $x *^i 1^S = x$ **if** $i \in I$ $x \in \mathbb{A}_i$ **for** $i\ x$
using *universes-subsets sum-one-def mult-map-def that* **by** *simp*
next
show $x \rightarrow^i x = 1^S$ **if** $i \in I$ $x \in \mathbb{A}_i$ **for** $i\ x$
using *universes-subsets imp-map-def sum-one-def that* **by** *simp*
next
show $x *^i (x \rightarrow^i y) = y *^i (y \rightarrow^i x)$
if $i \in I$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $i\ x\ y\ z$
using *divisibility universes-subsets imp-map-def mult-map-def that* **by** *simp*
next
show $x \rightarrow^i (y \rightarrow^i z) = (x *^i y) \rightarrow^i z$
if $i \in I$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $i\ x\ y\ z$
using *universes-subsets imp-map-def mult-map-def residuation that* **by** *simp*
next
show $x \leq^i y \vee y \leq^i x$ **if** $i \in I$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *total-order universes-subsets imp-map-def sum-one-def that* **by** *simp*
next
show $\nexists\ B\ C.$
$(\mathbb{A}_i = B \cup C) \wedge$
$(\{1^S\} = B \cap C) \wedge$
$(\exists\ y \in B.\ y \neq 1^S) \wedge$
$(\exists\ y \in C.\ y \neq 1^S) \wedge$
$(\mathit{hoop}\ B\ (*^i)\ (\rightarrow^i)\ 1^S) \wedge$
$(\mathit{hoop}\ C\ (*^i)\ (\rightarrow^i)\ 1^S) \wedge$
$(\forall\ x \in B - \{1^S\}.\ \forall\ y \in C.\ x *^i y = x) \wedge$
$(\forall\ x \in B - \{1^S\}.\ \forall\ y \in C.\ x \rightarrow^i y = 1^S) \wedge$
$(\forall\ x \in C.\ \forall\ y \in B.\ x \rightarrow^i y = y)$
if $i \in I$ **for** i
using *that Un-iff universes-one-fixed-hoops imp-map-def sum-one-def totally-ordered-one-fixed-hoop.one-fixed*
by *metis*
qed

lemma *same-uni* [*simp*]: *A-SUM.sum-univ* = *A*
  using *A-SUM.sum-univ-def universes-cover* **by** *auto*

lemma *floor-is-class*:
  assumes $a \in A-\{1^A\}$
  shows *A-SUM.floor* $a = \pi\ a$
proof −
  have $a \in \pi\ a \wedge \pi\ a \in I$
    using *index-set-def assms classes-not-empty* **by** *fastforce*
  then
  show *?thesis*
    using *same-uni A-SUM.floor-prop A-SUM.floor-unique UnCI assms universes-aux sum-one-def*
    by *metis*
qed

lemma *same-mult*:
  assumes $a \in A$ $b \in A$
  shows $a *^A b$ = *A-SUM.sum-mult a b*
proof −
  from *assms*
  consider (*1*) $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ *A-SUM.floor a* = *A-SUM.floor b*
    | (*2*) $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ *A-SUM.floor a* $<^I$ *A-SUM.floor b*
    | (*3*) $a \in A-\{1^A\}$ $b \in A-\{1^A\}$ *A-SUM.floor b* $<^I$ *A-SUM.floor a*
    | (*4*) $a = 1^A \vee b = 1^A$
    using *same-uni A-SUM.floor-prop fixed-points.cases sum-one-def trichotomy*
    by *metis*
  then
  show *?thesis*
  proof(*cases*)
    case *1*
    then
    show *?thesis*
      using *A-SUM.sum-mult.simps*(*1*) *sum-one-def mult-map-def* **by** *auto*
  next
    case *2*
    define *i j* **where** *i* = *A-SUM.floor a* **and** *j* = *A-SUM.floor b*
    then
    have $i \in I \wedge j \in I \wedge a \in (\mathbb{A}_i)-\{1^A\} \wedge b \in (\mathbb{A}_j)-\{1^A\}$
      using *2*(*1*,*2*) *A-SUM.floor-prop sum-one-def* **by** *auto*
    then
    have $a *^A b = a$
      using *2*(*3*) *different-indexes-mult i-def j-def* **by** *blast*
    moreover
    have *A-SUM.sum-mult a b* = *a*
      using *2 A-SUM.sum-mult.simps*(*2*) *sum-one-def* **by** *simp*
    ultimately
    show *?thesis*
      by *simp*

```
  next
    case 3
    define i j where i = A-SUM.floor a and j = A-SUM.floor b
    then
    have i ∈ I ∧ j ∈ I ∧ a ∈ (𝔸i)−{1A} ∧ b ∈ (𝔸j)−{1A}
      using 3(1,2) A-SUM.floor-prop sum-one-def by auto
    then
    have a *A b = b
      using 3(3) assms different-indexes-mult i-def j-def mult-comm by metis
    moreover
    have A-SUM.sum-mult a b = b
      using 3 A-SUM.sum-mult.simps(3) sum-one-def by simp
    ultimately
    show ?thesis
      by simp
  next
    case 4
    then
    show ?thesis
      using A-SUM.mult-neutr A-SUM.mult-neutr-2 assms sum-one-def by force
  qed
qed

lemma same-imp:
  assumes a ∈ A b ∈ A
  shows a →A b = A-SUM.sum-imp a b
proof −
  from assms
  consider (1) a ∈ A−{1A} b ∈ A−{1A} A-SUM.floor a = A-SUM.floor b
    | (2) a ∈ A−{1A} b ∈ A−{1A} A-SUM.floor a <I A-SUM.floor b
    | (3) a ∈ A−{1A} b ∈ A−{1A} A-SUM.floor b <I A-SUM.floor a
    | (4) a = 1A ∨ b = 1A
    using same-uni A-SUM.floor-prop fixed-points.cases sum-one-def trichotomy
    by metis
  then
  show ?thesis
  proof(cases)
    case 1
    then
    show ?thesis
      using A-SUM.sum-imp.simps(1) imp-map-def sum-one-def by auto
  next
    case 2
    define i j where i = A-SUM.floor a and j = A-SUM.floor b
    then
    have i ∈ I ∧ j ∈ I ∧ a ∈ (𝔸i)−{1A} ∧ b ∈ (𝔸j)−{1A}
      using 2(1,2) A-SUM.floor-prop sum-one-def by simp
    then
    have a →A b = 1A
```

using *2(3) different-indexes-imp-1 i-def j-def* **by** *blast*
moreover
have $A\text{-}SUM.sum\text{-}imp\ a\ b = 1^A$
using *2 A-SUM.sum-imp.simps(2) sum-one-def* **by** *simp*
ultimately
show *?thesis*
by *simp*
next
case *3*
define *i j* **where** $i = A\text{-}SUM.floor\ a$ **and** $j = A\text{-}SUM.floor\ b$
then
have $i \in I \wedge j \in I \wedge a \in (\mathbb{A}_i) - \{1^A\} \wedge b \in (\mathbb{A}_j) - \{1^A\}$
using *3(1,2) A-SUM.floor-prop sum-one-def* **by** *simp*
then
have $a \rightarrow^A b = b$
using *3(3) different-indexes-imp-2 i-def j-def* **by** *blast*
moreover
have $A\text{-}SUM.sum\text{-}imp\ a\ b = b$
using *3 A-SUM.sum-imp.simps(3) sum-one-def* **by** *auto*
ultimately
show *?thesis*
by *simp*
next
case *4*
then
show *?thesis*
using *A-SUM.imp-one-C A-SUM.imp-one-top assms imp-one-C imp-one-top sum-one-def*
by *force*
qed
qed

lemma *ordinal-sum-is-totally-ordered-hoop*:
totally-ordered-hoop A-SUM.sum-univ A-SUM.sum-mult A-SUM.sum-imp 1^S
proof
show $A\text{-}SUM.hoop\text{-}order\ x\ y \vee A\text{-}SUM.hoop\text{-}order\ y\ x$
if $x \in A\text{-}SUM.sum\text{-}univ$ $y \in A\text{-}SUM.sum\text{-}univ$ **for** *x y*
using *that A-SUM.hoop-order-def total-order hoop-order-def sum-one-def same-imp*
by *auto*
qed

theorem *totally-ordered-hoop-is-equal-to-ordinal-sum-of-tower-of-irr-hoops*:
shows *eq-universe*: $A = A\text{-}SUM.sum\text{-}univ$
and *eq-mult*: $x \in A \Longrightarrow y \in A \Longrightarrow x *^A y = A\text{-}SUM.sum\text{-}mult\ x\ y$
and *eq-imp*: $x \in A \Longrightarrow y \in A \Longrightarrow x \rightarrow^A y = A\text{-}SUM.sum\text{-}imp\ x\ y$
and *eq-one*: $1^A = 1^S$
proof
show $A \subseteq A\text{-}SUM.sum\text{-}univ$

by *simp*
next
show $A\text{-}SUM.sum\text{-}univ \subseteq A$
by *simp*
next
show $x *^A y = A\text{-}SUM.sum\text{-}mult\ x\ y$ **if** $x \in A$ $y \in A$ **for** x y
using *same-mult that* **by** *blast*
next
show $x \to^A y = A\text{-}SUM.sum\text{-}imp\ x\ y$ **if** $x \in A$ $y \in A$ **for** x y
using *same-imp that* **by** *blast*
next
show $1^A = 1^S$
using *sum-one-def* **by** *simp*
qed

4.5.9 Remarks on the nontrivial case

In the nontrivial case we have that every totally ordered hoop can be written as the ordinal sum of a tower of nontrivial irreducible hoops. The proof of this fact is almost immediate. By definition, $\mathbb{A}_{\pi 1 A} = \{1^A\}$ is the only trivial hoop in our tower. Moreover, $\mathbb{A}_{\pi a}$ is non-trivial for every $a \in A - \{1^A\}$. Given that $1^A \in \mathbb{A}_i$ for every $i \in I$ we can simply remove $\pi\ 1^A$ from I and obtain the desired result.

lemma *nontrivial-tower*:
assumes $\exists\ x \in A.\ x \neq 1^A$
shows
tower-of-nontrivial-irr-hoops $(I - \{\pi\ 1^A\})$ $(\leq^I)$ $(<^I)$ UNI_A MUL_A IMP_A 1^S
proof
show $I - \{\pi\ 1^A\} \neq \emptyset$
proof −
obtain a **where** $a \in A - \{1^A\}$
using *assms* **by** *blast*
then
have $\pi\ a \in I - \{\pi\ 1^A\}$
using *A-SUM.floor-prop class-not-one class-one floor-is-class sum-one-def* **by** *auto*
then
show *?thesis*
by *auto*
qed
next
show *reflp-on* $(I - \{\pi\ 1^A\})$ $(\leq^I)$
using *Diff-subset reflex reflp-on-subset* **by** *meson*
next
show *antisymp-on* $(I - \{\pi\ 1^A\})$ $(\leq^I)$
using *Diff-subset antisymm antisymp-on-subset* **by** *meson*
next
show *transp-on* $(I - \{\pi\ 1^A\})$ $(\leq^I)$

using *Diff-subset trans transp-on-subset* **by** *meson*
next
show $i <^I j = (i \leq^I j \wedge i \neq j)$ **if** $i \in I-\{\pi\ 1^A\}$ $j \in I-\{\pi\ 1^A\}$ **for** $i\ j$
using *index-order-strict-def* **by** *simp*
next
show *totalp-on* $(I-\{\pi\ 1^A\})$ $(\leq^I)$
using *Diff-subset total totalp-on-subset* **by** *meson*
next
show $(\mathbb{A}_i) \cap (\mathbb{A}_j) = \{1^S\}$ **if** $i \in I-\{\pi\ 1^A\}$ $j \in I-\{\pi\ 1^A\}$ $i \neq j$ **for** $i\ j$
using *A-SUM.almost-disjoint that* **by** *blast*
next
show $x *^i y \in \mathbb{A}_i$ **if** $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *A-SUM.floor-mult-closed that* **by** *blast*
next
show $x \rightarrow^i y \in \mathbb{A}_i$ **if** $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *A-SUM.floor-imp-closed that* **by** *blast*
next
show $1^S \in \mathbb{A}_i$ **if** $i \in I-\{\pi\ 1^A\}$ **for** i
using *universes-one-closed sum-one-def that* **by** *simp*
next
show $x *^i y = y *^i x$ **if** $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *universes-subsets mult-comm mult-map-def that* **by** *force*
next
show $x *^i (y *^i z) = (x *^i y) *^i z$
if $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $i\ x\ y\ z$
using *universes-subsets mult-assoc mult-map-def that* **by** *force*
next
show $x *^i 1^S = x$ **if** $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ **for** $i\ x$
using *universes-subsets sum-one-def mult-map-def that* **by** *force*
next
show $x \rightarrow^i x = 1^S$ **if** $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ **for** $i\ x$
using *universes-subsets imp-map-def sum-one-def that* **by** *force*
next
show $x *^i (x \rightarrow^i y) = y *^i (y \rightarrow^i x)$
if $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $i\ x\ y\ z$
using *divisibility universes-subsets imp-map-def mult-map-def that* **by** *auto*
next
show $x \rightarrow^i (y \rightarrow^i z) = (x *^i y) \rightarrow^i z$
if $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ $z \in \mathbb{A}_i$ **for** $i\ x\ y\ z$
using *universes-subsets imp-map-def mult-map-def residuation that* **by** *force*
next
show $x \leq^i y \vee y \leq^i x$ **if** $i \in I-\{\pi\ 1^A\}$ $x \in \mathbb{A}_i$ $y \in \mathbb{A}_i$ **for** $i\ x\ y$
using *DiffE total-order universes-subsets imp-map-def sum-one-def that* **by** *metis*
next
show $\nexists\ B\ C.$
$(\mathbb{A}_i = B \cup C) \wedge$
$(\{1^S\} = B \cap C) \wedge$
$(\exists\ y \in B.\ y \neq 1^S) \wedge$

($\exists\ y \in C.\ y \neq 1^S$) $\wedge$
(*hoop B* ($*^i$) ($\rightarrow^i$) 1^S) $\wedge$
(*hoop C* ($*^i$) ($\rightarrow^i$) 1^S) $\wedge$
($\forall\ x \in B-\{1^S\}.\ \forall\ y \in C.\ x *^i y = x$) $\wedge$
($\forall\ x \in B-\{1^S\}.\ \forall\ y \in C.\ x \rightarrow^i y = 1^S$) $\wedge$
($\forall\ x \in C.\ \forall\ y \in B.\ x \rightarrow^i y = y$)
if $i \in I-\{\pi\ 1^A\}$ **for** i
using *that Diff-iff Un-iff universes-one-fixed imp-map-def sum-one-def* **by** *metis*
next
show $\exists\ x \in \mathbb{A}_i.\ x \neq 1^S$ **if** $i \in I-\{\pi\ 1^A\}$ **for** i
using *universes-def indexes-class-of-elements indexes-not-empty that*
by *fastforce*
qed

lemma *ordinal-sum-of-nontrivial*:
assumes $\exists\ x \in A.\ x \neq 1^A$
shows *A-SUM.sum-univ* $= \{x.\ \exists\ i \in I-\{\pi\ 1^A\}.\ x \in \mathbb{A}_i\}$
proof
show *A-SUM.sum-univ* $\subseteq \{x.\ \exists\ i \in I-\{\pi\ 1^A\}.\ x \in \mathbb{A}_i\}$
proof
fix *a*
assume $a \in$ *A-SUM.sum-univ*
then
consider (*1*) $a \in A-\{1^A\}$
| (*2*) $a = 1^A$
by *auto*
then
show $a \in \{x.\ \exists\ i \in I-\{\pi\ 1^A\}.\ x \in \mathbb{A}_i\}$
proof(*cases*)
case *1*
then
obtain *i* **where** $i = \pi\ a$
by *simp*
then
have $a \in \mathbb{A}_i \wedge i \in I-\{\pi\ 1^A\}$
using *1 A-SUM.floor-prop class-not-one class-one floor-is-class sum-one-def*
by *auto*
then
show *?thesis*
by *blast*
next
case *2*
obtain *c* **where** $c \in A-\{1^A\}$
using *assms* **by** *blast*
then
obtain *i* **where** $i = \pi\ c$
by *simp*
then
have $a \in \mathbb{A}_i \wedge i \in I-\{\pi\ 1^A\}$

```
        using 2 A-SUM.floor-prop ‹c ∈ A−{1^A}› class-not-one class-one
              universes-one-closed floor-is-class sum-one-def
        by auto
      then
      show ?thesis
        by auto
    qed
  qed
next
  show {x. ∃ i ∈ I−{π 1^A}. x ∈ 𝔸_i} ⊆ A-SUM.sum-univ
    using universes-subsets by force
qed

end
```

4.5.10 Converse of main result

We show that the converse of the main result also holds, that is, the ordinal sum of a tower of irreducible hoops is a totally ordered hoop.

```
context tower-of-irr-hoops
begin

proposition ordinal-sum-of-tower-of-irr-hoops-is-totally-ordered-hoop:
  shows totally-ordered-hoop S (*^S) (→^S) 1^S
proof
  show hoop-order a b ∨ hoop-order b a if a ∈ S b ∈ S for a b
  proof −
    from that
    consider (1) a ∈ S−{1^S} b ∈ S−{1^S} floor a = floor b
      | (2) a ∈ S−{1^S} b ∈ S−{1^S} floor a <^I floor b ∨ floor b <^I floor a
      | (3) a = 1^S ∨ b = 1^S
      using floor.cases floor-prop trichotomy by metis
    then
    show hoop-order a b ∨ hoop-order b a
    proof(cases)
      case 1
      then
      have a ∈ 𝔸_(floor a) ∧ b ∈ 𝔸_(floor a)
        using 1 floor-prop by metis
      moreover
      have totally-ordered-hoop (𝔸_(floor a)) (*^(floor a)) (→^(floor a)) 1^S
        using 1(1) family-of-irr-hoops totally-ordered-irreducible-hoop.axioms(1)
              floor-prop
        by meson
      ultimately
      have a →^(floor a) b = 1^S ∨ b →^(floor a) a = 1^S
        using hoop.hoop-order-def totally-ordered-hoop.total-order
              totally-ordered-hoop-def
        by meson
```

```
    moreover
    have a →^S b = a →^floor a b ∧ b →^S a = b →^floor a a
      using 1 by auto
    ultimately
    show ?thesis
      using hoop-order-def by force
  next
    case 2
    then
    show ?thesis
      using sum-imp.simps(2) hoop-order-def by blast
  next
    case 3
    then
    show ?thesis
      using that ord-top by auto
  qed
 qed
qed

end

end
```

5 BL-chains

BL-chains generate the variety of BL-algebras, the algebraic counterpart of the Basic Fuzzy Logic (see [6]). As mentioned in the abstract, this formalization is based on the proof for BL-chains found in [5]. We define *BL−chain* and *bounded tower of irreducible hoops* and formalize the main result on that paper (Theorem 3.4).

```
theory BL-Chains
  imports Totally-Ordered-Hoops

begin
```

5.1 Definitions

```
locale bl-chain = totally-ordered-hoop +
  fixes zeroA :: 'a (0^A)
  assumes zero-closed: 0^A ∈ A
  assumes zero-first: x ∈ A ⟹ 0^A ≤^A x

locale bounded-tower-of-irr-hoops = tower-of-irr-hoops +
  fixes zeroI (0^I)
  fixes zeroS (0^S)
  assumes I-zero-closed : 0^I ∈ I
  and zero-first: i ∈ I ⟹ 0^I ≤^I i
```

and *first-zero-closed*: $0^S \in UNI\ 0^I$
and *first-bounded*: $x \in UNI\ 0^I \Longrightarrow IMP\ 0^I\ 0^S\ x = 1^S$
begin

abbreviation (*uni-zero*)
uni-zero :: $'b\ set$ ($\mathbb{A}_{0I}$)
where $\mathbb{A}_{0I} \equiv UNI\ 0^I$

abbreviation (*imp-zero*)
imp-zero :: $['b, 'b] \Rightarrow 'b$ $(((-)/ \rightarrow^{0I} / (-))\ [61,61]\ 60)$
where $x \rightarrow^{0I} y \equiv IMP\ 0^I\ x\ y$

end

context *bl-chain*
begin

5.2 First element of I

definition *zeroI* :: $'a\ set$ (0^I)
where $0^I = \pi\ 0^A$

lemma *I-zero-closed*: $0^I \in I$
using *index-set-def zeroI-def zero-closed* **by** *auto*

lemma *I-has-first-element*:
assumes $i \in I\ i \neq 0^I$
shows $0^I <^I i$
proof −
have $x \leq^A y$ **if** $i <^I 0^I\ x \in i\ y \in 0^I$ **for** $x\ y$
using *I-zero-closed assms*(*1*) *index-order-strict-def that* **by** *fastforce*
then
have $x \leq^A 0^A$ **if** $i <^I 0^I\ x \in i$ **for** x
using *classes-not-empty zeroI-def zero-closed that* **by** *simp*
moreover
have $0^A \leq^A x$ **if** $x \in i$ **for** x
using *assms*(*1*) *that in-mono indexes-subsets zero-first* **by** *meson*
ultimately
have $x = 0^A$ **if** $i <^I 0^I\ x \in i$ **for** x
using *assms*(*1*) *indexes-subsets ord-antisymm zero-closed that* **by** *blast*
moreover
have $0^A \in 0^I$
using *classes-not-empty zeroI-def zero-closed* **by** *simp*
ultimately
have $i \cap 0^I \neq \emptyset$ **if** $i <^I 0^I$
using *assms*(*1*) *indexes-not-empty that* **by** *force*
moreover
have $i <^I 0^I \vee 0^I <^I i$
using *I-zero-closed assms trichotomy* **by** *auto*

ultimately
show *?thesis*
using *I-zero-closed assms(1) indexes-disjoint* **by** *auto*
qed

5.3 Main result for BL-chains

definition *zeroS* :: *'a* (0^S)
where $0^S = 0^A$

abbreviation (*uniA-zero*)
uniA-zero :: *'a set* (($\mathbb{A}_{0I}$))
where $\mathbb{A}_{0I} \equiv UNI_A\ 0^I$

abbreviation (*impA-zero-xy*)
impA-zero-xy :: [*'a*, *'a*] ⇒ *'a* (((-)/ $\rightarrow^{0I}$ / (-)) [61, 61] 60)
where $x \rightarrow^{0I} y \equiv IMP_A\ 0^I\ x\ y$

lemma *tower-is-bounded*:
shows *bounded-tower-of-irr-hoops I* ($\leq^I$) ($<^I$) UNI_A MUL_A IMP_A 1^S 0^I 0^S
proof
show $0^I \in I$
using *I-zero-closed* **by** *simp*
next
show $0^I \leq^I i$ **if** $i \in I$ **for** i
using *I-has-first-element index-ord-reflex index-order-strict-def that* **by** *blast*
next
show $0^S \in \mathbb{A}_{0I}$
using *classes-not-empty universes-def zeroI-def zeroS-def zero-closed* **by** *simp*
next
show $0^S \rightarrow^{0I} x = 1^S$ **if** $x \in \mathbb{A}_{0I}$ **for** x
using *I-zero-closed universes-subsets hoop-order-def imp-map-def sum-one-def zeroS-def zero-first that*
by *simp*
qed

lemma *ordinal-sum-is-bl-totally-ordered*:
shows *bl-chain A-SUM.sum-univ A-SUM.sum-mult A-SUM.sum-imp* 1^S 0^S
proof
show *A-SUM.hoop-order x y* ∨ *A-SUM.hoop-order y x*
if $x \in$ *A-SUM.sum-univ* $y \in$ *A-SUM.sum-univ* **for** *x y*
using *ordinal-sum-is-totally-ordered-hoop totally-ordered-hoop.total-order that*
by *meson*
next
show $0^S \in$ *A-SUM.sum-univ*
using *zeroS-def zero-closed* **by** *simp*
next
show *A-SUM.hoop-order* 0^S *x* **if** $x \in$ *A-SUM.sum-univ* **for** *x*
using *A-SUM.hoop-order-def eq-imp hoop-order-def sum-one-def zeroS-def zero-closed*

zero-first that
  by *simp*
qed

theorem *bl-chain-is-equal-to-ordinal-sum-of-bounded-tower-of-irr-hoops*:
 shows *eq-universe*: $A = A\text{-}SUM.sum\text{-}univ$
 and *eq-mult*: $x \in A \Longrightarrow y \in A \Longrightarrow x *^A y = A\text{-}SUM.sum\text{-}mult\ x\ y$
 and *eq-imp*: $x \in A \Longrightarrow y \in A \Longrightarrow x \to^A y = A\text{-}SUM.sum\text{-}imp\ x\ y$
 and *eq-zero*: $0^A = 0^S$
 and *eq-one*: $1^A = 1^S$
proof
 show $A \subseteq A\text{-}SUM.sum\text{-}univ$
  by *auto*
next
 show $A\text{-}SUM.sum\text{-}univ \subseteq A$
  by *auto*
next
 show $x *^A y = A\text{-}SUM.sum\text{-}mult\ x\ y$ **if** $x \in A\ y \in A$ **for** $x\ y$
  using *eq-mult that* **by** *blast*
next
 show $x \to^A y = A\text{-}SUM.sum\text{-}imp\ x\ y$ **if** $x \in A\ y \in A$ **for** $x\ y$
  using *eq-imp that* **by** *blast*
next
 show $0^A = 0^S$
  using *zeroS-def* **by** *simp*
next
 show $1^A = 1^S$
  using *sum-one-def* **by** *simp*
qed

end

5.4 Converse of main result for BL-chains

context *bounded-tower-of-irr-hoops*
begin

We show that the converse of the main result holds if $0^S \neq 1^S$. If $0^S = 1^S$ then the converse may not be true. For example, take a trivial hoop A and an arbitrary not bounded Wajsberg hoop B such that $A \cap B = \{1\}$. The ordinal sum of both hoops is equal to B and therefore not bounded.

proposition *ordinal-sum-of-bounded-tower-of-irr-hoops-is-bl-chain*:
 assumes $0^S \neq 1^S$
 shows *bl-chain* $S\ (*^S)\ (\to^S)\ 1^S\ 0^S$
proof
 show *hoop-order* $a\ b \lor$ *hoop-order* $b\ a$ **if** $a \in S\ b \in S$ **for** $a\ b$
 proof −
  from *that*
  consider $(1)\ a \in S - \{1^S\}\ b \in S - \{1^S\}$ *floor* $a =$ *floor* b

| (*2*) $a \in S-\{1^S\}$ $b \in S-\{1^S\}$ *floor* $a <^I$ *floor* $b \vee$ *floor* $b <^I$ *floor* a
| (*3*) $a = 1^S \vee b = 1^S$
using *floor.cases floor-prop trichotomy* **by** *metis*
then
show *?thesis*
proof(*cases*)
case *1*
then
have $a \in \mathbb{A}_{floor\ a} \wedge b \in \mathbb{A}_{floor\ a}$
using *1 floor-prop* **by** *metis*
moreover
have *totally-ordered-hoop* $(\mathbb{A}_{floor\ a})$ $(*^{floor\ a})$ $(\rightarrow^{floor\ a})$ 1^S
using *1*(*1*) *family-of-irr-hoops totally-ordered-irreducible-hoop.axioms*(*1*)
floor-prop
by *meson*
ultimately
have $a \rightarrow^{floor\ a} b = 1^S \vee b \rightarrow^{floor\ a} a = 1^S$
using *hoop.hoop-order-def totally-ordered-hoop.total-order*
totally-ordered-hoop-def
by *meson*
moreover
have $a \rightarrow^S b = a \rightarrow^{floor\ a} b \wedge b \rightarrow^S a = b \rightarrow^{floor\ a} a$
using *1* **by** *auto*
ultimately
show *?thesis*
using *hoop-order-def* **by** *force*
next
case *2*
then
show *?thesis*
using *sum-imp.simps*(*2*) *hoop-order-def* **by** *blast*
next
case *3*
then
show *?thesis*
using *that ord-top* **by** *auto*
qed
qed
next
show $0^S \in S$
using *first-zero-closed I-zero-closed sum-subsets* **by** *auto*
next
show *hoop-order* 0^S a **if** $a \in S$ **for** a
proof −
have *zero-dom*: $0^S \in \mathbb{A}_{0I} \wedge 0^S \in S-\{1^S\}$
using *I-zero-closed sum-subsets assms first-zero-closed* **by** *blast*
moreover
have *floor* $0^S \leq^I$ *floor* x **if** $0^S \in S-\{1^S\}$ $x \in S-\{1^S\}$ **for** x
using *I-zero-closed floor-prop floor-unique that*(*2*) *zero-dom zero-first*

by *metis*
ultimately
have *floor* $0^S \le^I$ *floor* x **if** $x \in S-\{1^S\}$ **for** x
using *that* **by** *blast*
then
consider (1) $0^S \in S-\{1^S\}$ $a \in S-\{1^S\}$ *floor* 0^S = *floor* a
| (2) $0^S \in S-\{1^S\}$ $a \in S-\{1^S\}$ *floor* $0^S <^I$ *floor* a
| (3) $a = 1^S$
using ‹$a \in S$› *floor.cases floor-prop strict-order-equiv-not-converse trichotomy zero-dom*
by *metis*
then
show *hoop-order* 0^S a
proof(*cases*)
case *1*
then
have $0^S \in \mathbb{A}_{0I} \wedge a \in \mathbb{A}_{0I}$
using *I-zero-closed first-zero-closed floor-prop floor-unique* **by** *metis*
then
have $0^S \to^S a = 0^S \to^{0I} a \wedge 0^S \to^{0I} a = 1^S$
using *1 I-zero-closed sum-imp.simps*(*1*) *first-bounded floor-prop floor-unique*
by *metis*
then
show *?thesis*
using *hoop-order-def* **by** *blast*
next
case *2*
then
show *?thesis*
using *sum-imp.simps*(*2*,*5*) *hoop-order-def* **by** *meson*
next
case *3*
then
show *?thesis*
using *ord-top zero-dom* **by** *auto*
qed
qed
qed

end

end

References

[1] P. Agliano and F. Montagna. Varieties of BL-algebras I: general properties. *Journal of Pure and Applied Algebra*, 181(2):105–129, 2003.

[2] W. J. Blok and M. A. Ferreirim. On the structure of hoops. *Algebra Universalis*, 43(2):233–257, 2000.

[3] B. Bosbach. Komplementäre Halbgruppen. Axiomatik und Arithmetik. *Fundamenta Mathematicae*, 64:257–287, 1969.

[4] J. R. Büchi and T. M. Owens. Complemented monoids and hoops. *unpublished manuscript*, 1975.

[5] M. Busaniche. Decomposition of BL-chains. *Algebra Universalis*, 52(4):519–525, 2005.

[6] P. Hájek. *Metamathematics of Fuzzy Logic*. Kluwer Academic Publishers, Dordrecht, Boston and London, 1998.